Supplementary Mathematica file.

CONTENTS:

Part 0: “Housekeeping”, definitions of functions, matrices and replacement rules that will be used throughout the file.

Part 1: Probabilities of identity by descent (related to Appendix B).

We compare the formulas calculated by hand to the ones obtained numerically for special structures, to check that the formulas are correct (they are).

Part 2: Expected Frequencies Functions (related to Appendix A and formulas in the main text)

- Computation of E[X] for each life-cycle, written as factor * (R B - C),
- Signs of the B term, and of the competition terms,
- Effects of m on E[X] for each life-cycle,
- Qualitative effects of m for Moran DB.

Part 3: Expected Frequencies Functions, generic method (i.e. not restricted to the type of subdivided population that we consider in the main text) -> related to the Supplementary Figures

- We simplify the formulas obtained by hand by replacing the dispersal (d) and interaction (e) graphs by their formulas in subdivided populations.
- We compare the formulas calculated by hand to the ones obtained numerically for special structures, to check that the formulas are correct (they are).
- We export the formulas to R for further use.

Please note:

- In this file, the mutation bias is sometimes denoted by p (as in Tarnita and Taylor 2014 and Debarre 2017), instead of v as in the manuscript.
  The letter v was chosen in the manuscript because p was sometimes mistaken by others as average frequency of altruists in the population ($\overline{X}$ in the manuscript)... But this file was written before the change, and it is too complicated to change every instance of “p”.

- In this file, the number of demes is denoted by d instead of N_D in the article.

- Make sure to change `pathtosave` with the path to the folder containing the codes.

Before doing anything, clean the memory

```
Clear[Evaluate[Context[] <> "*"]]
```

Set path to folder where outputs should be saved (otherwise it is the default Mathematica one)

```
pathtosave = "";
```

0) Generalities - Initializations

Some functions

Function to turn P (expected state of pairs of sites) into Q (probabilities of identity by descent)

In[•]:= $$\text{PtoQ[P\_] :=} \frac{\text{P} - \text{p}^2}{\text{p}\ (1 - \text{p})} \text{ // FullSimplify;}$$

Delta function

In[•]:=
```
Delta[x_] := If[x == 0, 1, 0]
```

Define graphs for numerical evaluation

Dispersal and Interaction Graphs

Island model, dispersal graph, generic

N = 12, 4 demes of 3 individuals

In[•]:=
```
G12generic =

  ( dself  din    din    dout   dout   dout   dout   dout   dout   dout   dout   dout
    din    dself  din    dout   dout   dout   dout   dout   dout   dout   dout   dout
    din    din    dself  dout   dout   dout   dout   dout   dout   dout   dout   dout
    dout   dout   dout   dself  din    din    dout   dout   dout   dout   dout   dout
    dout   dout   dout   din    dself  din    dout   dout   dout   dout   dout   dout
    dout   dout   dout   din    din    dself  dout   dout   dout   dout   dout   dout
    dout   dout   dout   dout   dout   dout   dself  din    din    dout   dout   dout
    dout   dout   dout   dout   dout   dout   din    dself  din    dout   dout   dout
    dout   dout   dout   dout   dout   dout   din    din    dself  dout   dout   dout
    dout   dout   dout   dout   dout   dout   dout   dout   dout   dself  din    din
    dout   dout   dout   dout   dout   dout   dout   dout   dout   din    dself  din
    dout   dout   dout   dout   dout   dout   dout   dout   dout   din    din    dself

   ) ;

Nin = 3;
Ndemes = 4;
```

N = 10, 2 demes of 5 individuals

In[•]:=
$$\text{G10generic} = \begin{pmatrix} \text{dself} & \text{din} & \text{din} & \text{din} & \text{din} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} \\ \text{din} & \text{dself} & \text{din} & \text{din} & \text{din} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} \\ \text{din} & \text{din} & \text{dself} & \text{din} & \text{din} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} \\ \text{din} & \text{din} & \text{din} & \text{dself} & \text{din} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} \\ \text{din} & \text{din} & \text{din} & \text{din} & \text{dself} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} \\ \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dself} & \text{din} & \text{din} & \text{din} & \text{din} \\ \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{din} & \text{dself} & \text{din} & \text{din} & \text{din} \\ \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{din} & \text{din} & \text{dself} & \text{din} & \text{din} \\ \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{din} & \text{din} & \text{din} & \text{dself} & \text{din} \\ \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{dout} & \text{din} & \text{din} & \text{din} & \text{din} & \text{dself} \end{pmatrix};$$

Island model, interaction graph, generic

In[•]:=
```
GE12generic = G12generic /. {dself → eself, dout → eout, din → ein};
GE10generic = G10generic /. {dself → eself, dout → eout, din → ein};
```

Formulas for d and e

Replacements for the generic dispersal probabilities, depending on whether there is self-replacement or not

In[•]:=
$$\text{noselfreplacement} = \left\{\text{dself} \to 0,\ \text{din} \to \frac{1-m}{n-1},\ \text{dout} \to \frac{m}{d\,n - n}\right\};$$
$$\text{withselfreplacement} = \left\{\text{dself} \to \frac{1-m}{n},\ \text{din} \to \frac{1-m}{n},\ \text{dout} \to \frac{m}{d\,n - n}\right\};$$

Replacements for the generic interaction probabilities, depending on whether there is self - interaction or not

In[•]:=
$$\text{groupnoself} = \left\{\text{eself} \to 0,\ \text{ein} \to \frac{1}{n-1},\ \text{eout} \to 0\right\};$$
$$\text{groupwithself} = \left\{\text{eself} \to \frac{1}{n},\ \text{ein} \to \frac{1}{n},\ \text{eout} \to 0\right\};$$

We can even assume that there are a proportion g of interactions outside of the group

In[•]:=
$$\text{widewithself} = \left\{\text{eself} \to \frac{1-g}{n},\ \text{ein} \to \frac{1-g}{n},\ \text{eout} \to \frac{g}{n\,d - n}\right\};$$
$$\text{widenoself} = \left\{\text{eself} \to 0,\ \text{ein} \to \frac{1-g}{n-1},\ \text{eout} \to \frac{g}{n\,d - n}\right\};$$

Combine these using Idself and Ieself, indicator variables for whether there is dispersal/interaction with self

In[•]:=
$$\text{genericde} = \left\{\text{dself} \to \text{Idself}\,\frac{1-m}{n},\ \text{din} \to (1-m)\left(\frac{1}{n-1} - \text{Idself}\,\frac{1}{n\,(n-1)}\right),\ \text{dout} \to \frac{m}{n\,d-n},\ (* \right.$$
$$\left. *)\ \text{eself} \to \text{Ieself}\,\frac{1-g}{n},\ \text{ein} \to (1-g)\left(\frac{1}{n-1} - \text{Ieself}\,\frac{1}{n\,(n-1)}\right),\ \text{eout} \to \frac{g}{n\,d-n}\right\};$$

Quick check

```
In[•]:= eself + (n - 1) ein + (n d - n) eout /. genericde // Simplify
        dself + (n - 1) din + (n d - n) dout /. genericde // Simplify
Out[•]= 1
Out[•]= 1
```

Probabilities of identity by descent matrices

Generic Q matrix corresponding to the populations defined above

N = 12

```
In[•]:= Q12generic =
```

$$\begin{pmatrix} 1 & Qin & Qin & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout \\ Qin & 1 & Qin & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout \\ Qin & Qin & 1 & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout \\ Qout & Qout & Qout & 1 & Qin & Qin & Qout & Qout & Qout & Qout & Qout & Qout \\ Qout & Qout & Qout & Qin & 1 & Qin & Qout & Qout & Qout & Qout & Qout & Qout \\ Qout & Qout & Qout & Qin & Qin & 1 & Qout & Qout & Qout & Qout & Qout & Qout \\ Qout & Qout & Qout & Qout & Qout & Qout & 1 & Qin & Qin & Qout & Qout & Qout \\ Qout & Qout & Qout & Qout & Qout & Qout & Qin & 1 & Qin & Qout & Qout & Qout \\ Qout & Qout & Qout & Qout & Qout & Qout & Qin & Qin & 1 & Qout & Qout & Qout \\ Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & 1 & Qin & Qin \\ Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qin & 1 & Qin \\ Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qout & Qin & Qin & 1 \end{pmatrix};$$

N = 10

```
In[•]:= Q10generic =
```

$$\begin{pmatrix} 1 & Qin & Qin & Qin & Qin & Qout & Qout & Qout & Qout & Qout \\ Qin & 1 & Qin & Qin & Qin & Qout & Qout & Qout & Qout & Qout \\ Qin & Qin & 1 & Qin & Qin & Qout & Qout & Qout & Qout & Qout \\ Qin & Qin & Qin & 1 & Qin & Qout & Qout & Qout & Qout & Qout \\ Qin & Qin & Qin & Qin & 1 & Qout & Qout & Qout & Qout & Qout \\ Qout & Qout & Qout & Qout & Qout & 1 & Qin & Qin & Qin & Qin \\ Qout & Qout & Qout & Qout & Qout & Qin & 1 & Qin & Qin & Qin \\ Qout & Qout & Qout & Qout & Qout & Qin & Qin & 1 & Qin & Qin \\ Qout & Qout & Qout & Qout & Qout & Qin & Qin & Qin & 1 & Qin \\ Qout & Qout & Qout & Qout & Qout & Qin & Qin & Qin & Qin & 1 \end{pmatrix};$$

1) Probabilities of identity by descent (Q)

Moran

Simplify QinM and QoutM

See Appendix B2 for calculation details on how Qself, Qin and Qout were obtained using a formula presented in the appendix of Debarre 2017 JTB for “2D graphs”. Here we just copy these formulas.

$$\text{QselfM2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{\mu} + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(n-1) + \frac{1}{1-(1-\mu)\,\left(1-m-\frac{m}{d-1}\right)}\,(d-1) + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(d-1)\,(n-1)\right);$$

$$\text{QinM2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{\mu} + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(-1) + \frac{1}{1-(1-\mu)\,\left(1-m-\frac{m}{d-1}\right)}\,(d-1) + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(d-1)\,(-1)\right);$$

$$\text{QoutM2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{\mu} + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(-1) + \frac{1}{1-(1-\mu)\,\left(1-m-\frac{m}{d-1}\right)}\,(-1) + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\right);$$

Find λ using Qself == 1

```
theλM = λ /. Solve[QselfM2 == 1, λ][[1]] // FullSimplify
```

$$\left(n\,(1+\text{din}+\text{dself}\,(-1+\mu)-\text{din}\,\mu)\,(-d\,m+\mu+d\,(-1+m)\,\mu)\right) \Big/ \left(m\,(-1+\mu)\,(1+\text{din}-\text{dself}+(-\text{din}+\text{dself}+d\,(-1+n))\,\mu) + (-1+d)\,\mu\,(-1+\text{dself}+\text{din}\,(-1+\mu)+\mu-(\text{dself}+n)\,\mu)\right)$$

Replace λ in the equations for Qin and Qout

```
QinM = QinM2 /. λ → theλM // FullSimplify
QoutM = QoutM2 /. λ → theλM // FullSimplify
```

$$\left((-1+\mu)\,((-1+d)\,(1+\text{din}-\text{dself})\,\mu+m\,(1+\text{din}-\text{dself}-(d+\text{din}-\text{dself})\,\mu))\right) \Big/ \left(m\,(-1+\mu)\,(1+\text{din}-\text{dself}+(-\text{din}+\text{dself}+d\,(-1+n))\,\mu) + (-1+d)\,\mu\,(-1+\text{dself}+\text{din}\,(-1+\mu)+\mu-(\text{dself}+n)\,\mu)\right)$$

$$\left(m\,(-1+\mu)\,(1+\text{din}+\text{dself}\,(-1+\mu)-\text{din}\,\mu)\right) \Big/ \left(m\,(-1+\mu)\,(1+\text{din}-\text{dself}+(-\text{din}+\text{dself}+d\,(-1+n))\,\mu) + (-1+d)\,\mu\,(-1+\text{dself}+\text{din}\,(-1+\mu)+\mu-(\text{dself}+n)\,\mu)\right)$$

Check numerically

Here we evaluate the probabilities of identity by descent numerically, using the recursion formula (“eqs” in the function below), with specific graphs.

```
NGetQM[G_, N_, graphdegree_, p_, μ_] := Module[{QT, eqs, vars, sols, QTs}, (*
  G is the dispersal graph,
  N is the size of the population,
  graphdegree is the degree of the graph (=1 in a subdivided population),
  p is the mutation biais,
  μ is the mutation probability.
  *)

  (* Initialize the QT matrix *)
  Do[Qi,j = 0; Qi,j =., {i, 1, N}, {j, 1, N}];
  QT = Table[Qi,j, {i, 1, N}, {j, 1, N}];
  Do[QT[[i, i]] = 1, {i, 1, N}]; (* Qi,i= 1 *)
  Do[QT[[i, j]] = Qj,i, {j, 1, N - 1}, {i, j + 1, N}]; (* Because Q is symmetric *)

  eqs = Flatten[
    Table[Qi,j == (1 - μ)/(2 graphdegree) (Sum[G[[l, j]] QT[[l, i]] + G[[l, i]] QT[[l, j]], {l, 1, N}]),
     {i, 1, N - 1}, {j, i + 1, N}]];

  vars = Flatten[Table[Qi,j, {i, 1, N - 1}, {j, i + 1, N}]];
  sols = NSolve[eqs, vars];

  QTs = QT /. sols[[1]];
  QTs]
```

This function compares the numerical version to the analytical one, for specific graph structures. If the numerical values are the same, we are fine! (and we are, otherwise there would not be a paper)

```
prs =.;
CheckQM[prs_, dvalues_] := Module[{NG, NQin, NQout, NQinMatrix},
  NG = ToExpression["G" <> ToString[d n /. prs] <> "generic"] /. dvalues /. prs;
  NQinMatrix = NGetQM[NG, n d /. prs, 1, 0.5, μ /. prs];
  NQin = NQinMatrix[[1, 2]];
  NQout = NQinMatrix[[1, n d /. prs]];
  Print[{{"Qin", NQin, QinM /. dvalues /. prs},
     {"Qout", NQout, QoutM /. dvalues /. prs}} // Transpose // MatrixForm]
 ]
```

Check for the population of size 12

```
CheckQM[{m → 0.2, d → 4, n → 3, μ → 0.2}, withselfreplacement]
CheckQM[{m → 0.2, d → 4, n → 3, μ → 0.2}, noselfreplacement]
```

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.407643 & 0.127389 \\ 0.407643 & 0.127389 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.503735 & 0.140875 \\ 0.503735 & 0.140875 \end{pmatrix}$$

Check for the population of size 10

```
CheckQM[{m → 0.2, d → 2, n → 5, μ → 0.2}, withselfreplacement]
CheckQM[{m → 0.2, d → 2, n → 5, μ → 0.2}, noselfreplacement]
```

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.329897 & 0.206186 \\ 0.329897 & 0.206186 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.3762 & 0.222649 \\ 0.3762 & 0.222649 \end{pmatrix}$$

Particular cases

Equations with self replacement (dself = din)

```
QinMs = QinM /. withselfreplacement // FullSimplify
QoutMs = QoutM /. withselfreplacement // FullSimplify
```

$$-\frac{(-1+\mu)\left(\mu - d\,\mu + m\left(-1+d\,\mu\right)\right)}{-\left(-1+d\right)\mu\,(1+(-1+n)\,\mu) + m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)}$$

$$\frac{m\,(-1+\mu)}{-\left(-1+d\right)\mu\,(1+(-1+n)\,\mu) + m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)}$$

Simplify the way the equations are written (human - friendly versions), and check that the formulas remain correct

```
QinMs - ((1 - μ) (d μ (1 - m) - μ + m)) /
    ((d - 1) μ (1 + (n - 1) μ) + m (1 - μ) (1 + d (n - 1) μ)) // FullSimplify
```

0

```
QoutMs - (m (1 - μ)) / ((d - 1) μ (1 + (n - 1) μ) + m (1 - μ) (1 + d (n - 1) μ)) // FullSimplify
```

0

Check limit behavior:

First infinite population, then zero mutation
vs.
First zero mutation, then infinite population

```
Limit[QoutMs, d → ∞] // FullSimplify
Limit[Limit[QoutMs, μ → 0], d → ∞] // FullSimplify
```

0

1

```
Limit[QinMs, d → ∞] // FullSimplify
% /. μ → 0 // FullSimplify
```

$$\frac{-1+m+\mu-m\,\mu}{-1+m\,(-1+n)\,(-1+\mu)+\mu-n\,\mu}$$

$$\frac{1-m}{1+m\,(-1+n)}$$

```
Limit[QinMs, μ → 0]
Limit[%, d → ∞] // FullSimplify
```

1

1

```
Series[QinMs, {μ, 0, 1}]
```

$1 - d\,n\,\mu + O[\mu]^2$

Equations without Self - replacement (dself = 0)

```
QinMw = QinM /. noselfreplacement // FullSimplify
QoutMw = QoutM /. noselfreplacement // FullSimplify
```

$$\frac{(-1+\mu)\left(m^2(-1+\mu)+(-1+d)\,n\,\mu+m\,(n-d\,n\,\mu)\right)}{m^2(-1+\mu)^2-(-1+d)\,n\,\mu\,(1+(-2+n)\,\mu)+m\,n\,(-1+\mu)\,(1+d\,(-2+n)\,\mu)}$$

$$\frac{m\,(n+m\,(-1+\mu)-\mu)\,(-1+\mu)}{m^2(-1+\mu)^2-(-1+d)\,n\,\mu\,(1+(-2+n)\,\mu)+m\,n\,(-1+\mu)\,(1+d\,(-2+n)\,\mu)}$$

Simplify the way they are written

```
QinMw - ((1 - μ) (d n μ (1 - m) + (m - μ) n - m² (1 - μ))) /
    (+ (d - 1) n μ (1 + (n - 2) μ) + m n (1 - μ) (1 + d (n - 2) μ) - m² (1 - μ)²) // FullSimplify
```

0

```
QoutMw - (m (n + m (-1 + μ) - μ) (1 - μ)) /
    (+ (d - 1) n μ (1 + (n - 2) μ) + m n (1 - μ) (1 + d (n - 2) μ) - m² (1 - μ)²) // FullSimplify
```

0

Limit behavior

```
Limit[QinMw, d → ∞] // FullSimplify
Limit[%, μ → 0] // FullSimplify
```

$$\frac{-1+m+\mu-m\,\mu}{-1+m\,(-2+n)\,(-1+\mu)-(-2+n)\,\mu}$$

$$\frac{1-m}{1+m\,(-2+n)}$$

```
Limit[QoutMw, d → ∞]
Limit[QoutMw, μ → 0]
```

0

1

Wright - Fisher

The structure of this part is the same as for the Moran version above, so comments are lighter here.

Simplify QinM and QoutM

See Appendix B2 for details on how Qself, Qin and Qout were obtained using a formula presented in the appendix of Debarre 2017 JTB for "2D graphs".

In[•]:=
$$\mathtt{QselfWF2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{1-(1-\mu)^2}+\frac{1}{1-(1-\mu)^2\,(\mathtt{dself}-\mathtt{din})^2}\,(n-1)\,d+\frac{1}{1-(1-\mu)^2\left(1-m-\frac{m}{d-1}\right)^2}\,(d-1)\right);$$

$$\mathtt{QinWF2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{1-(1-\mu)^2}-\frac{1}{1-(1-\mu)^2\,(\mathtt{dself}-\mathtt{din})^2}\,d+\frac{1}{1-(1-\mu)^2\left(1-m-\frac{m}{d-1}\right)^2}\,(d-1)\right);$$

$$\mathtt{QoutWF2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{1-(1-\mu)^2}-\frac{1}{1-(1-\mu)^2\left(1-m-\frac{m}{d-1}\right)^2}\right);$$

Find λ using Qself == 1

In[•]:=
```
λWF = λ /. Solve[QselfWF2 == 1, λ][[1]]
```

Out[•]=
$$\frac{d\,n}{\left(\frac{1}{1-(1-\mu)^2}+\frac{d\,(-1+n)}{1-(-\mathtt{din}+\mathtt{dself})^2\,(1-\mu)^2}+\frac{-1+d}{1-\left(1-m-\frac{m}{-1+d}\right)^2\,(1-\mu)^2}\right)\mu}$$

Replace λ in the equations

In[•]:=
```
QinWF = QinWF2 /. λ → λWF // FullSimplify
QoutWF = QoutWF2 /. λ → λWF // FullSimplify
```

Out[•]=
$$\frac{-\frac{d}{1-(\mathtt{din}-\mathtt{dself})^2\,(-1+\mu)^2}+\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{d\,(-1+n)}{1-(\mathtt{din}-\mathtt{dself})^2\,(-1+\mu)^2}+\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

Out[•]=
$$\frac{-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{d\,(-1+n)}{1-(\mathtt{din}-\mathtt{dself})^2\,(-1+\mu)^2}+\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

Check numerically

```
In[•]:= NGetQWF[G_, N_, graphdegree_, p_, μ_] := Module[{QT, eqs, vars, sols, QTs}, (*
    G is the dispersal graph,
    N is the size of the population,
    graphdegree is the degree of the graph (=1 in a subdivided pop),
    p is the mutation biais,
    μ is the mutation probability.
    *)

    (* Initialize the QT matrix *)
    Do[Q_{i,j} = 0; Q_{i,j} =., {i, 1, N}, {j, 1, N}];
    QT = Table[Q_{i,j}, {i, 1, N}, {j, 1, N}];
    Do[QT[[i, i]] = 1, {i, 1, N}]; (* Q_{i,i}=1 *)
    Do[QT[[i, j]] = Q_{j,i}, {j, 1, N - 1}, {i, j + 1, N}]; (* Because Q is symmetric *)

    eqs = Flatten[
      Table[Q_{i,j} == (1 - μ)^2 / graphdegree^2 (Sum[G[[l, j]] G[[k, i]] QT[[k, l]], {l, 1, N}, {k, 1, N}]),
       {i, 1, N - 1}, {j, i + 1, N}]];

    vars = Flatten[Table[Q_{i,j}, {i, 1, N - 1}, {j, i + 1, N}]];
    sols = NSolve[eqs, vars];

    QTs = QT /. sols[[1]];
    QTs]

In[•]:= prs =.;
   CheckQWF[prs_, dvalues_] := Module[{NG, NQin, NQout, NQinMatrix},
     NG = ToExpression["G" <> ToString[d n /. prs] <> "generic"] /. dvalues /. prs;
     NQinMatrix = NGetQWF[NG, n d /. prs, 1, 0.5, μ /. prs];
     NQin = NQinMatrix[[1, 2]];
     NQout = NQinMatrix[[1, n d /. prs]];
     Print[{{"Qin", NQin, QinWF /. dvalues /. prs},
         {"Qout", NQout, QoutWF /. dvalues /. prs}} // Transpose // MatrixForm]
    ]

In[•]:= CheckQWF[{m → 0.2, d → 4, n → 3, μ → 0.2}, withselfreplacement]
   CheckQWF[{m → 0.2, d → 4, n → 3, μ → 0.2}, noselfreplacement]
   CheckQWF[{m → 0.2, d → 2, n → 5, μ → 0.2}, withselfreplacement]
   CheckQWF[{m → 0.2, d → 2, n → 5, μ → 0.2}, noselfreplacement]
```

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.218352 & 0.0816154 \\ 0.218352 & 0.0816154 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.178044 & 0.0770357 \\ 0.178044 & 0.0770357 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.17199 & 0.122413 \\ 0.17199 & 0.122413 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.164772 & 0.120319 \\ 0.164772 & 0.120319 \end{pmatrix}$$

Particular cases

With Self Replacement

In[•]:= `QinWFs = QinWF /. withselfreplacement // FullSimplify`

Out[•]=
$$\frac{-d + \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}{d(-1+n) + \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}$$

Simplify the way it is written

In[•]:=
$$\text{M1} = \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}};$$

$$\text{M2} = \frac{1}{2\mu-\mu^2};$$

$$\text{qinwfs} = \frac{-d + \text{M1} + \text{M2}}{(n-1)\,d + \text{M1} + \text{M2}};$$

`QinWFs - qinwfs // FullSimplify`

Out[•]= 0

In[•]:= `QoutWFs = QoutWF /. withselfreplacement // FullSimplify`

Out[•]=
$$\frac{-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}{d(-1+n) + \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}$$

Simplify the way it is written

In[•]:=
$$\text{qoutwfs} = \frac{\frac{-1}{d-1}\,\text{M1} + \text{M2}}{d\,(n-1) + \text{M1} + \text{M2}} \;\text{// FullSimplify};$$

`qoutwfs - QoutWFs // FullSimplify`

Out[•]= 0

Limit behavior

In[•]:=
```
Limit[QinWFs, μ → 0]
Limit[QoutWFs, μ → 0]
```

Out[•]= 1

Out[•]= 1

In[•]:=
```
Limit[QinWFs, d → ∞] // FullSimplify
Limit[%, μ → 0] // FullSimplify
```

Out[•]=
$$-\frac{(-1+m)^2\,(-1+\mu)^2}{-1-2\,m\,(-1+n)\,(-1+\mu)^2+m^2\,(-1+n)\,(-1+\mu)^2+(-1+n)\,(-2+\mu)\,\mu}$$

Out[•]=
$$-\frac{(-1+m)^2}{-1+(-2+m)\,m\,(-1+n)}$$

In[•]:=
```
Limit[QoutWFs, d → ∞] // FullSimplify
```

Out[•]= 0

Comparison to Cockerham and Weir 1987

Cockerham and Weir’s β

In[•]:=
$$dd = \left(1 - m\,\frac{nbdemes}{nbdemes - 1}\right)^2;$$
$$\rho = (1-\mu)^2;$$

Update deme size, which is 2 N in their paper, to N, to adapt the result to a haploid population

In[•]:=
```
β = ρ dd / (demesize (1 - ρ dd) + ρ dd) // FullSimplify
ρ =.; dd =.;
```

Out[•]=
$$\frac{(1+(-1+m)\,nbdemes)^2\,(-1+\mu)^2}{(-1+nbdemes)^2\left(demesize\left(1-\frac{(1+(-1+m)\,nbdemes)^2\,(-1+\mu)^2}{(-1+nbdemes)^2}\right)+\frac{(1+(-1+m)\,nbdemes)^2\,(-1+\mu)^2}{(-1+nbdemes)^2}\right)}$$

For us:

In[•]:=
$$my\beta = \frac{QinWFs - QoutWFs}{1 - QoutWFs}\ \text{// FullSimplify}$$

Out[•]=
$$-\left(\left((1+d\,(-1+m))^2\,(-1+\mu)^2\right)\Big/\left(-1+d\,(2-d+2\,m\,(-1+n)+d\,(-2+m)\,m\,(-1+n))+2\,\mu-2\,(d\,(2+d\,(-1+m))\,(-1+m)\,(-1+n)+n)\,\mu+(-1+n)\,(\mu+d\,(-1+m)\,\mu)^2\right)\right)$$

Compare the two -> same !

In[•]:=
```
myβ - β /. {nbdemes → d, demesize → n} /. {din → 1 / n, dself → 1 / n} // FullSimplify
```

Out[•]= 0

In[•]:=
```
β =.; myβ =.;
```

Without Self Replacement

In[∘]:=
```
QinWFw = QinWF /. noselfreplacement // FullSimplify
```

Out[∘]=
$$\frac{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2\,(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

In[∘]:=
```
QoutWFw = QoutWF /. noselfreplacement // FullSimplify
```

Out[∘]=
$$\frac{-\frac{1}{1-\frac{\left(1+d\,(-1+m)\right)^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

Export to R the Q results

Rewrite the Greek letters

```
GreekTerms = {ω → sel, μ → mut}
```

{ω → sel, μ → mut}

Common parts to all functions

```
FunctionPartB = " <- function(p, sel, mut, m, g, n, d, Idself, Ieself){
## Arguments:
#  p    mutation bias
#  sel  intensity of selection
#  mut  mutation probability
#  m    emigration probability
#  g    proportion of interactions
      out of the group (interaction equivalent of m)
#  n    deme size
#  d    number of demes
#  Idself whether reproduction in site where the parent is
#  Ieself whether interactions with oneself
return(
";
FunctionPartE = ")
}";
```

Function to translate Mathematica to R

```
ToRForm[x_] := ToString[x /. GreekTerms // CForm]
```

Do it for Q

```
RtxtQinM = "QinM " <> FunctionPartB <>
    ToRForm[QinM /. genericde // FullSimplify] <> FunctionPartE;
RtxtQoutM = "QoutM " <> FunctionPartB <>
    ToRForm[QoutM /. genericde // FullSimplify] <> FunctionPartE;
RtxtQinWF = "QinWF " <> FunctionPartB <>
    ToRForm[QinWF /. genericde // FullSimplify] <> FunctionPartE;
RtxtQoutWF = "QoutWF " <> FunctionPartB <>
    ToRForm[QoutWF /. genericde // FullSimplify] <> FunctionPartE;
```

Define Power function in R

```
PowerDef = "Power <- function(a,b) return(a^b)";
```

Combine all texts

```
Rtxt = PowerDef <> "

" <> RtxtQinM <> "

" <> RtxtQoutM <> "

" <> RtxtQinWF <> "

" <> RtxtQoutWF;
```

Export to txt file (Mathematica did not want R)

```
Export[pathtosave <> "Mathematica/analyticsQ.txt", Rtxt];
```

Convert the file extension to R

```
cmd = "mv " <> pathtosave <> "Mathematica/analyticsQ.txt " <>
    pathtosave <> "Mathematica/analyticsQ.R";
```

```
Get["!" <> cmd]
```

2) Expected frequency of altruists in a subdivided population

Initializations

Conditions on the parameters

```
assumpts =
   {n > 1 && d > 2 && μ > 0 && μ < 1 && m > 0 && m < 1 && b > 0 && c > 0 && b > c && ν > 0 && ν < 1};
```

FullSimplify with conditions on the parameters

```
AF[x_] := Assuming[assumpts, FullSimplify[x]]
```

Relatedness

```
R2M = (QinM - QoutM)/(1 - QoutM) /. genericde /. {Idself → 1} // AF
R2WF = (QinWF - QoutWF)/(1 - QoutWF) /. genericde /. {Idself → 1} // AF
```

$$-\frac{(1+d(-1+m))(-1+\mu)}{1+(-1+n)\mu+d(-1+m(-1+n)(-1+\mu)+\mu-n\mu)}$$

$$-\left(\left((1+d(-1+m))^2(-1+\mu)^2\right)\Big/\left(-1+d\left(2(1+m(-1+n))+d\left(-1+(-2+m)m(-1+n)\right)\right)+2\mu-2\left(d\left(2+d(-1+m)\right)(-1+m)(-1+n)+n\right)\mu+\left(1+d(-1+m)\right)^2(-1+n)\mu^2\right)\right)$$

Parameters

Set parameter values for numerical examples

```
prm = {d → 15, n → 4, b → 15, c → 1, δ → 0.05, ν → 0.45};
```

Derivatives of the expected frequency of altruists

Definitions

Fitness derivative: decomposition
Notation: s=self, i=deme-mate who is not the focal

```
dW = dWs + (n - 1) dWi R2;
```

Further decomposition with fecundities:

```
dWs = dWfs (-c) + b dWfi;
dWi = dWfs b/(n - 1) + dWfi (-c) + (n - 2)/(n - 1) b dWfi;
```

Values for each life-cycle, obtained by taking derivatives in table S2”

Moran Death - Birth

$$\text{dWfsDB} = (1-\mu)\,\frac{1}{n\,d}\left(1-\frac{(1-m)^2}{n}-\frac{m^2}{n\,(d-1)}\right);$$

$$\text{dWfiDB} = (1-\mu)\,\frac{1}{n\,d}\left(-\frac{(1-m)^2}{n}-\frac{m^2}{n\,(d-1)}\right);$$

Combine with the formula for dW

```
dWDB = dW /. {dWfs → dWfsDB, dWfi → dWfiDB} // AF;
BDB = (n - 1) dWi /. {dWfs → dWfsDB, dWfi → dWfiDB} // AF;
CDB = -dWs /. {dWfs → dWfsDB, dWfi → dWfiDB} // AF;
```

$$dEXDB = n\,d\,\frac{(1-Qout)}{\mu}\,\nu\,(1-\nu)\,dW \;/.\; \{dWfs \to dWfsDB,\; dWfi \to dWfiDB\} \;/.\; R2 \to R2M \;/.\; Qout \to QoutM \;/.\; \{dself \to din\} \;/.\; din \to \frac{1-m}{n};$$

Moran Birth - Death

$$dWfsBD = (1-\mu)\left(\frac{1}{n\,d} - \frac{1}{(n\,d)^2}\right) - \left(\frac{(1-m)}{n\,n\,d} - \frac{1}{(n\,d)^2}\right);$$

$$dWfiBD = (1-\mu)\left(-\frac{1}{(n\,d)^2}\right) - \left(\frac{(1-m)}{n\,n\,d} - \frac{1}{(n\,d)^2}\right);$$

Combine with dW

$$dEXBD = n\,d\,\frac{(1-Qout)}{\mu}\,\nu\,(1-\nu)\,dW \;/.\; \{dWfs \to dWfsBD,\; dWfi \to dWfiBD\} \;/.\; R2 \to R2M \;/.\; Qout \to QoutM \;/.\; \{dself \to din\} \;/.\; din \to \frac{1-m}{n};$$

```
BBD = (n - 1) dWi /. {dWfs → dWfsBD, dWfi → dWfiBD} // AF;
CBD = -dWs /. {dWfs → dWfsBD, dWfi → dWfiBD} // AF;
```

Wright - Fisher

$$dWfsWF = (1-\mu)\left(1 - \frac{(1-m)^2}{n} - \frac{m^2}{n\,(d-1)}\right);$$

$$dWfiWF = (1-\mu)\left(-\frac{(1-m)^2}{n} - \frac{m^2}{n\,(d-1)}\right);$$

Combine with dW

$$dEXWF = \frac{(1-Qout)}{\mu}\,\nu\,(1-\nu)\,dW \;/.\; \{dWfs \to dWfsWF,\; dWfi \to dWfiWF\} \;/.\; R2 \to R2WF \;/.\; Qout \to QoutWF \;/.\; \{dself \to din\} \;/.\; din \to \frac{1-m}{n};$$

```
BWF = (n - 1) dWi /. {dWfs → dWfsWF, dWfi → dWfiWF} // AF;
CWF = -dWs /. {dWfs → dWfsWF, dWfi → dWfiWF} // AF;
```

Checking signs

Death-Birth, whole B term

```
AF[Reduce[BDB > 0, m]]
```

$$\frac{-1 + d - \sqrt{\frac{(-1+d)\,(b+c\,(-1+n)+b\,(-1+d)\,n)}{(b-c)\,(-1+n)}}}{d} < m < \frac{-1 + d + \sqrt{\frac{(-1+d)\,(b+c\,(-1+n)+b\,(-1+d)\,n)}{(b-c)\,(-1+n)}}}{d}$$

and this is true when m<$\frac{d-1}{d}$.

```
AF[Reduce[CDB > 0, m]]
```
True

Competition terms

Check for BD competition

Competition term in the BD life-cycle

$$\text{BDcomp} = \frac{\mu}{n\, d} - \frac{1-m}{n};$$

It increases with m (because its derivative is >0)

```
D[BDcomp, m] // AF
```
$$\frac{1}{n}$$

It is negative when m=0 (because μ<d)

```
BDcomp /. m → 0 // AF
```
$$\frac{-d+\mu}{d\, n}$$

And it is still negative when we reach mc

$$\text{BDcomp /. m} \to \frac{d-1}{d} \text{ // AF}$$

$$\frac{-1+\mu}{d\, n}$$

So it is negative for all values of m that we consider, and increases with m.
-> Competition is reduced as m increases (the absolute value of the competition term decreases)

Check for DB & WF competition

The competition term is negative:

$$\text{DBcomp} = -(1-\mu)\left(\frac{(1-m)^2}{n} + \frac{m^2}{n\,(d-1)}\right);$$

How does it change with m?
Derivative with respect to m

```
dcomp = D[DBcomp, m] // AF
```
$$\frac{2\,(1+d\,(-1+m))\,(-1+\mu)}{(-1+d)\, n}$$

Derivative changes sign at the critical m value

```
Solve[dcomp == 0, m]
```
$$\left\{\left\{m \to \frac{-1+d}{d}\right\}\right\}$$

Second derivative is negative -> derivative decreases

-> derivative is positive for m<1-1/d (and indeed it is for m->0)
-> competition term increases with m, BUT...

```
D[dcomp, m] // AF
```

$$\frac{2\,d\,(-1+\mu)}{(-1+d)\,n}$$

```
dcomp /. m → 0 // AF
```

$$\frac{2-2\,\mu}{n}$$

... BUT competition term is negative, so competition is reduced as m increases

Changes with m

Birth - Death

Derivative with respect to m

```
ddEXBD = D[dEXBD, m] // AF
```

$$\frac{(-1+d)^2\,\mu\,(c+c\,d\,n-c\,\mu+b\,(-1+(1+d\,n\,(-2+\mu))\,\mu))\,(-1+\nu)\,\nu}{d\,((-1+d)\,\mu\,(1+(-1+n)\,\mu)-m\,(-1+\mu)\,(1+d\,(-1+n)\,\mu))^2}$$

There are no roots -> monotonic change with m

```
Solve[ddEXBD == 0, m] // AF
```

{}

How does the sign of the derivative change with μ?

```
Solve[ddEXBD == 0, μ] // AF
```

$$\left\{\{\mu\to 0\},\ \left\{\mu\to -\frac{b-c-2\,b\,d\,n+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n}\right\},\ \left\{\mu\to \frac{c+b\,(-1+2\,d\,n)+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n}\right\}\right\}$$

Select the admissible solution

```
μcBD = μ /. %[[2]] // AF;
```

Value with our parameters

```
μcBD /. prm // N
```

0.0260991

How is it affected by changes in d or n? (they play the same role)

The derivative is positive : the threshold increases with d and n

```
D[μcBD, d] // AF
```

$$\frac{\sqrt{\frac{b-c}{b-c+4\,b\,d^2\,n^2}}\,\left(b-c+\sqrt{(b-c)\,\left(b-c+4\,b\,d^2\,n^2\right)}\right)}{2\,b\,d^2\,n}$$

The maximum value of the threshold is

```
Limit[μcBD, d → ∞] // AF
% /. {b → 15, c → 1} // N
```

$$1-\sqrt{1-\frac{c}{b}}$$

0.0339082

–> The critical value of μ above which E[X] increases with m is

$$\mu cBD = 1-\frac{b-c+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n};$$

this threshold increases with d and n to reach a maximum value $\mu cBDmax = 1-\sqrt{1-\frac{c}{b}}$

Death - Birth

Derivative with respect to m

```
ddEXDB = D[dEXDB, m] // AF
```

$$\Big(\big((-1+\mu)\,\big(-c\,\big(-d\,m^2+\mu+\big(n+d\,\big(m\,(2+m)-2\,(1+n)-d\,(-1+m)\,(1+m\,(-1+n)+n)\big)\big)\,\mu+(1+d\,(-1+m))^2\,(-1+n)\,\mu^2\big)+b\,\big(-d\,m^2+\mu+d\,\big(-2+m\,(2+m)+d\,(1+m\,(-2+m-m\,n))\big)\,\mu+\big(-(1+d\,(-1+m))^2+\big(2+2\,d\,(-2+m)+d^2\,(2+(-2+m)\,m)\big)\,n\big)\,\mu^2\big)\big)\,(-1+\nu)\,\nu\Big)\Big/\Big(\big((-1+d)\,\mu\,(1+(-1+n)\,\mu)-m\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)\big)^2$$

The change is not monotonic with m:

```
Solve[ddEXDB == 0, m] // AF
```

$$\Big\{\Big\{m\to\Big((-1+d)\,\Big((b-c)\,d\,\mu\,(1+(-1+n)\,\mu)-\sqrt{\big(-(b-c)\,d\,\mu\,\big(c\,(1+n)+c\,\mu\,(-2+d\,n^2+\mu-n\,\mu)+b\,\big(-1+\mu\,\big(2-\mu+n\,\big(2\,(-1+\mu)+d\,(-1+(-1+n)\,(-2+\mu)\,\mu)\big)\big)\big)\big)\big)}\Big)\Big)\Big/\Big((b-c)\,d\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)\Big)\Big\},\Big\{m\to\Big((-1+d)\,\Big((b-c)\,d\,\mu\,(1+(-1+n)\,\mu)+\sqrt{\big(-(b-c)\,d\,\mu\,\big(c\,(1+n)+c\,\mu\,(-2+d\,n^2+\mu-n\,\mu)+b\,\big(-1+\mu\,\big(2-\mu+n\,\big(2\,(-1+\mu)+d\,(-1+(-1+n)\,(-2+\mu)\,\mu)\big)\big)\big)\big)\big)}\Big)\Big)\Big/\Big((b-c)\,d\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)\Big)\Big\}\Big\}$$

Select admissible solution (denominator <0)

```
mcDB = m /. %[[1]]
```

$$\frac{(-1+d)\left((b-c)\,d\,\mu\,(1+(-1+n)\,\mu)-\sqrt{-(b-c)\,d\,\mu\left(c\,(1+n)+c\,\mu\left(-2+d\,n^2+\mu-n\,\mu\right)+b\left(-1+\mu\left(2-\mu+n\left(2\,(-1+\mu)+d\left(-1+(-1+n)\,(-2+\mu)\,\mu\right)\right)\right)\right)\right)}\right)}{(b-c)\,d\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)}$$

We need to know whether this is a max or a min...
But the solutions are a bit complicated, so let' s focus on what happens at m -> 0

```
ddEXDB0 = ddEXDB /. m → 0 // AF
```

$$\frac{(-1+\mu)\left(b+b\,(-1+2\,n)\,\mu-c\,(1+n+(-1+n)\,\mu)\right)(-1+\nu)\,\nu}{\mu\,(1+(-1+n)\,\mu)^2}$$

How does this change with μ?

```
dddEXDB0 = D[ddEXDB0, μ] // AF
```

$$-\frac{1}{\mu^2\,(1+(-1+n)\,\mu)^3}\left(c\,(1+n)-c\,(-1+n)\,\mu\left(-3\,(1+n)+3\,\mu+(-1+n)\,\mu^2\right)+b\left(-1+\mu\left(3-3\,n\,(-1+\mu)^2+(-3+\mu)\,\mu+2\,n^2\,(-2+\mu)\,\mu\right)\right)\right)(-1+\nu)\,\nu$$

```
AF[Reduce[ddEXDB0 > 0, μ]]
AF[Solve[ddEXDB0 == 0, μ]]
```

$$c+c\,n+b\,\mu+c\,n\,\mu<b+c\,\mu+2\,b\,n\,\mu\ ||\ b\geq c+c\,n$$

$$\left\{\{\mu\to 1\},\ \left\{\mu\to\frac{-b+c+c\,n}{c-c\,n+b\,(-1+2\,n)}\right\}\right\}$$

-> E[X] is a non - monotonic function of m;

The initial (i.e., for m -> 0) increase of E[X] with m depends on the values of b and μ :
if b > c (n + 1), E[X] initially increases with m ;
otherwise, E[X] initially increases with m if $\mu > \frac{-b+c+c\,n}{c-c\,n+b\,(-1+2\,n)}$;

Then, a maximum is reached at

$$m=\frac{(-1+d)\left((b-c)\,d\,\mu\,(1+(-1+n)\,\mu)-\sqrt{-(b-c)\,d\,\mu\left(c\,(1+n)+c\,\mu\left(-2+d\,n^2+\mu-n\,\mu\right)+b\left(-1+\mu\left(2-\mu+n\left(2\,(-1+\mu)+d\left(-1+(-1+n)\,(-2+\mu)\,\mu\right)\right)\right)\right)\right)}\right)}{(b-c)\,d\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)};$$

Wright - Fisher

Derivative with respect to m

```
ddEXWF = D[dEXWF, m] // AF
```

$$\frac{2\,(-1+d)^3\,(1+d\,(-1+m))\,n\,(-2+\mu)^2\,(-1+\mu)\,\mu\,\left(c+b\,(-2+\mu)\,\mu\right)(-1+\nu)\,\nu}{\left((-1+d)^2\,(-2+\mu)\,\mu\,\left(-1+(-1+n)\,(-2+\mu)\,\mu\right)-2\,(-1+d)\,m\,(-1+\mu)^2\left(-1+d\,(-1+n)\,(-2+\mu)\,\mu\right)+d\,m^2\,(-1+\mu)^2\left(-1+d\,(-1+n)\,(-2+\mu)\,\mu\right)\right)^2}$$

An extremum is reached at the maximum possible emigration value:

```
Solve[ddEXWF == 0, m] // AF
```

$$\left\{\left\{m \to \frac{-1+d}{d}\right\}\right\}$$

Whether it is a min or max depends on μ; let's consider for simplicity m->0

```
ddEXWF0 = ddEXWF /. m → 0 // AF
```

$$-\frac{2\, n\, (-1+\mu)\, \left(c+b\, \left(-2+\mu\right)\, \mu\right)\, (-1+\nu)\, \nu}{\mu\, \left(-1+(-1+n)\, \left(-2+\mu\right)\, \mu\right)^2}$$

```
AF[Reduce[ddEXWF0 > 0, μ]]
```

$$\sqrt{1-\frac{c}{b}}+\mu > 1$$

->E[X] is a monotonic function of m for 0 < m < 1 - 1/d,

and it is an increasing function for μ>1 - $\sqrt{1-\frac{c}{b}}$

Qualitative effect of m on E[X] in the Moran DB life - cycle

We have already checked that BDB>0

```
CBratio = CDB/BDB // AF
```

$$-\frac{\left(b-c\right)\, \left(-1+d\, (-1+m)^2+2\, m\right)+c\, \left(-1+d\right)\, n}{-c\, \left(-1+d\, (-1+m)^2+2\, m\right)\, (-1+n)+b\, \left(1-d\, (-1+m)^2+2\, m\, (-1+n)+d\, \left(-2+m\right)\, m\, n\right)}$$

Values of R and C/B for m -> 0
we have R<= C/B, == when μ->0

```
Limit[R2M, m → 0] // AF
Limit[%, μ → 0]
```

$$\frac{1-\mu}{1+(-1+n)\, \mu}$$

1

```
Limit[CBratio, m → 0] // AF
```

1

At the other extreme:

```
Limit[R2M, m → 1 - 1/d] // AF
```

0

```
Limit[CBratio, m → 1 - 1/d] // AF
```

$$\frac{b+c\, \left(-1+d\, n\right)}{b+c\, (-1+n)+b\, \left(-1+d\right)\, n}$$

Threshold emigration value to have R>C/B

```
mDBqualit = m /. Solve[R2M BDB - CDB == 0, m] // AF
```

$$\Big\{\frac{1}{2\,(b-c)\,d}\Big(-c\,(-2+d+d\,n)+c\,d\,(-1+n)\,\mu+b\,(-2+d+d\,\mu)-\sqrt{\left(-4\,(b-c)\,(-1+d)\,d\,(b+c\,(-1+n))\,\mu+(b\,(-2+d+d\,\mu)+c\,(2-d\,(1+n)+d\,(-1+n)\,\mu))^2\right)}\Big),$$
$$\frac{1}{2\,(b-c)\,d}\Big(-c\,(-2+d+d\,n)+c\,d\,(-1+n)\,\mu+b\,(-2+d+d\,\mu)+\sqrt{\left(-4\,(b-c)\,(-1+d)\,d\,(b+c\,(-1+n))\,\mu+(b\,(-2+d+d\,\mu)+c\,(2-d\,(1+n)+d\,(-1+n)\,\mu))^2\right)}\Big)\Big\}$$

3) Expected Frequency Equations - Generic method

Here we use the methodology presented in Débarre 2017 JTB, and similar terminology and decomposition of E[X].
The derivation is much more tedious... but it allows us to consider other kinds of subdivided populations:
- we do not specify the values of dself, din and dout (for instance we can have dself =0, no replacement of the parent by the offspring),
- we do not specify the values of eself, ein and eout (for instance, we can have eout proportional to dout, i.e. not restrict social interaction to deme-mates).

Formulas for the different life-cycles

The formulas for each term is obtained by hand, by replacing the dispersal and interaction graphs by their formulas in a subdivided population, from the equations given in Appendix B1. In some cases (e.g., βI for the Moran DB life-cycle), there is a large number of cases to consider when unpacking the sums.
D corresponds to direct / primary effects,
I corresponds to indirect / secondary effects.

Moran, Birth-Death

β

```
βBDD = (1 - μ) (eself + (n - 1) ein Qin + (n d - n) eout Qout);
```

```
βBDI = dself eself + (n - 1) din ein + (n d - n) dout eout(*
    *) + (n - 1) (din eself + dself ein + (n - 2) din ein + (n d - n) dout eout) Qin (*
    *) + (n d - n) (dself eout + (n - 1) din eout +
        dout eself + (n - 1) dout ein + (n d - 2 n) dout eout) Qout(*
    *) - μ/(n d) (1 + (n - 1) Qin + (n d - n) Qout) (eself + (n - 1) ein + (n d - n) eout);

factorBD = p (1 - p)/μ;

βBD = factorBD (βBDD - βBDI);
```

γ

```
γBDD = 1 - μ;

γBDI = dself + (n - 1) din Qin + (n d - n) dout Qout - μ/(n d) (1 + (n - 1) Qin + (n d - n) Qout);

γBD = factorBD (γBDD - γBDI) // FullSimplify
```

$$\frac{1}{d\,n\,\mu}(-1+p)\,p\,\big(d\,n\,(-1+\text{dself}+\text{din}\,(-1+n)\,\text{Qin}+(-1+d)\,\text{dout}\,n\,\text{Qout})+(-1+\text{Qin}+n\,(d-\text{Qin}+\text{Qout}-d\,\text{Qout}))\,\mu\big)$$

Moran, Death-Birth

β

```
βDBD = βBDD
```

$$(\text{eself}+\text{ein}\,(-1+n)\,\text{Qin}+\text{eout}\,(-n+d\,n)\,\text{Qout})\,(1-\mu)$$

```
βDBI = (1 - μ) (1 * dself dself eself * 1(*
      02*) + (n - 1) dself din ein * 1(*
      03*) + (n d - n) dself dout eout * 1(*
      04*) + (n - 1) dself dself ein Qin(*
      05*) + (n - 1) dself din eself Qin(*
      06*) + (n - 1) (n - 2) dself din ein Qin(*
      07*) + (n - 1) (n d - n) dself dout eout Qin(*
      08*) + (n d - n) dself dself eout Qout(*
      09*) + (n d - n) (n - 1) dself din eout Qout(*
      10*) + (n d - n) dself dout eself Qout(*
      11*) + (n d - n) (n - 1) dself dout ein Qout(*
      12*) + (n d - n) (n d - 2 n) dself dout eout Qout(*
      13*) + (n - 1) din dself eself Qin(*
      14*) + (n - 1) din din ein Qin(*
      15*) + (n - 1) (n - 2) din din ein Qin (*
      16*) + (n - 1) (n d - n) din dout eout Qin(*
      17*) + (n - 1) din dself ein (*
      18*) + (n - 1) din din eself(*
```

```
19*) + (n - 1) (n - 2) din din ein(*
20*) + (n - 1) (n d - n) din dout eout(*
21*) + (n - 1) (n - 2) din dself ein Qin(*
22*) + (n - 1) (n - 2) din din ein Qin(*
23*) + (n - 1) (n - 2) din din eself Qin(*
24*) + (n - 1) (n - 2) (n - 3) din din ein Qin(*
25*) + (n - 1) (n - 2) (n d - n) din dout eout Qin(*
26*) + (n - 1) (n d - n) din dself eout Qout(*
27*) + (n - 1) (n d - n) din din eout Qout(*
28*) + (n - 1) (n d - n) (n - 2) din din eout Qout(*
29*) + (n - 1) (n d - n) din dout eself Qout(*
30*) + (n - 1) (n d - n) (n - 1) din dout ein Qout(*
31*) + (n - 1) (n d - n) (n d - 2 n) din dout eout Qout(*
32*) + (n d - n) dout dself eself Qout(*
33*) + (n d - n) (n - 1) dout din ein Qout(*
34*) + (n d - n) dout dout eout Qout(*
35*) + (n d - n) (n - 1) dout dout eout Qout(*
36*) + (n d - n) (n d - 2 n) dout dout eout Qout(*
37*) + (n d - n) (n - 1) dout dself ein Qout(*
38*) + (n d - n) (n - 1) dout din eself Qout(*
39*) + (n d - n) (n - 1) (n - 2) dout din ein Qout(*
40*) + (n d  - n) (n - 1) dout dout eout Qout(*
41*) + (n d - n) (n - 1) (n - 1) dout dout eout Qout(*
42*) + (n d - n) (n - 1) (n d - 2 n) dout dout eout Qout(*
43*) + (n d - n) dout dself eout (*
44*) + (n d - n) (n - 1) dout din eout(*
45*) + (n d - n) dout dout eself(*
46*) + (n d - n) (n - 1) dout dout ein(*
47*) + (n d - n) (n d - 2 n) dout dout eout(*
48*) + (n d - n) (n - 1) dout dself eout Qin(*
49*) + (n d - n) (n - 1) (n - 1) dout din eout Qin(*
50*) + (n d - n) (n - 1) dout dout ein Qin(*
51*) + (n d - n) (n - 1) dout dout eself Qin(*
52*) + (n d - n) (n - 1) (n - 2) dout dout ein Qin(*
53*) + (n d - n) (n - 1) (n d - 2 n) dout dout eout Qin(*
54*) + (n d - n) (n d - 2 n) dout dself eout Qout(*
55*) + (n d - n) (n d - 2 n) (n - 1) dout din eout Qout(*
56*) + (n d - n) (n d - 2 n) dout dout eout Qout(*
57*) + (n d - n) (n d - 2 n) (n - 1) dout dout eout Qout(*
58*) + (n d - n) (n d - 2 n) dout dout eself Qout(*
59*) + (n d - n) (n d - 2 n) (n - 1) dout dout ein Qout(*
60*) + (n d - n) (n d - 2 n) (n d - 3 n) dout dout eout Qout);

factorDB = factorBD
```

$$\frac{(1-p)\,p}{\mu}$$

```
βDB = factorDB (βDBD - βDBI) // FullSimplify
```

$$\frac{1}{\mu}(1-p)\,p\,\Big(\text{eself}+\text{ein}\,(-1+n)\,\text{Qin}+(-1+d)\,\text{eout}\,n\,\text{Qout}-\text{dself}^2\,\big(\text{eself}+\text{ein}\,(-1+n)\,\text{Qin}+(-1+d)\,\text{eout}\,n\,\text{Qout}\big)-\text{din}^2\,(-1+n)\,\big(\text{eself}+\text{eself}\,(-2+n)\,\text{Qin}+\text{ein}\,(-2+n+(3+(-3+n)\,n)\,\text{Qin})+(-1+d)\,\text{eout}\,(-1+n)\,n\,\text{Qout}\big)-(-1+d)\,\text{dout}^2\,n\,\big((\text{eself}+\text{ein}\,(-1+n)+(-2+d)\,\text{eout}\,n)\,(1+(-1+n)\,\text{Qin})+n\,((-2+d)\,\text{eself}+(-2+d)\,\text{ein}\,(-1+n)+(3+(-3+d)\,d)\,\text{eout}\,n)\,\text{Qout}\big)-2\,(-1+d)\,\text{dout}\,\text{dself}\,n\,\big((\text{eself}+\text{ein}\,(-1+n))\,\text{Qout}+\text{eout}\,(1+(-1+n)\,\text{Qin}+(-2+d)\,n\,\text{Qout})\big)-2\,\text{din}\,(-1+n)\,\big(\text{dself}\,(\text{ein}+\text{eself}\,\text{Qin}+\text{ein}\,(-2+n)\,\text{Qin}+(-1+d)\,\text{eout}\,n\,\text{Qout})+(-1+d)\,\text{dout}\,n\,(\text{eout}+\text{eout}\,(-1+n)\,\text{Qin}+(\text{eself}+\text{ein}\,(-1+n)+(-2+d)\,\text{eout}\,n)\,\text{Qout})\big)\Big)\,(1-\mu)$$

γ

```
γDBD = γBDD
```

$1-\mu$

```
γDBI = (1 - μ) (dself^2 + (n - 1) din^2 + (n d - n) dout^2 (*
    *) + (n - 1) (dself din + din dself + (n - 2) din din + (n d - n) dout dout) Qin (*
    *) + (n d - n) (dself dout + (n - 1) din dout +
      dout dself + (n - 1) dout din + (n d - 2 n) dout dout) Qout);
```

```
γDB = factorDB (γDBD - γDBI)
```

$$\frac{1}{\mu}(1-p)\,p\,\Big(1-\big(\text{dself}^2+\text{din}^2\,(-1+n)+\text{dout}^2\,(-n+d\,n)+(-1+n)\,(2\,\text{din}\,\text{dself}+\text{din}^2\,(-2+n)+\text{dout}^2\,(-n+d\,n))\,\text{Qin}+(-n+d\,n)\,(2\,\text{dout}\,\text{dself}+2\,\text{din}\,\text{dout}\,(-1+n)+\text{dout}^2\,(-2\,n+d\,n))\,\text{Qout}\big)\,(1-\mu)-\mu\Big)$$

Wright - Fisher

The formulas are the same as the Moran DB life-cycle, only the probabilities of identity by descent Q will differ.

β

```
βWFI = βDBI;
βWFD = βDBD;
βWF = βDB;
```

γ

```
γWF = γDB;
γWFD = γDBD;
γWFI = γDBI;
```

Check the results numerically

We use generic equations valid for any life-cycle and any graph, and adapt them to our life-cycles and to a subdivided population. We compare the numerical results to the ones obtained with the equations written above.

Full functions, any life-cycle

β and γ were calculated by hand - these are generic equations value for any life-cycle and any regular graph.
(see Appendix B1 for details, and Debarre 2017 JTB for even further details)

```
GetBeta[sBf_, Df_, G_, GE_, Qmat_, N_, graphdegree_, Bstar_] :=
 Module[{part1, part2, factor},
  factor = p (1 - p) / (μ N Bstar);
  part1 = Sum[((1 - μ) sBf[G, N, graphdegree, j, l] - Df[G, N, graphdegree, j, l]) *
     GE[[k, l]] * Qmat[[j, k]], {j, 1, N}, {k, 1, N}, {l, 1, N}];
  factor * part1]

GetGamma[sBf_, Df_, G_, GE_, Qmat_, N_, graphdegree_, Bstar_] :=
 Module[{part1, part2, factor},
  factor = p (1 - p) / (μ N Bstar);
  part1 = Sum[((1 - μ) sBf[G, N, graphdegree, j, k] - Df[G, N, graphdegree, j, k]) *
     Qmat[[j, k]], {j, 1, N}, {k, 1, N}];
  factor *
   part1]
```

Moran DB

Define δB and δD

graphdegree is the degree of the graph, here equal to 1

```
sBfDB[G_, N_, graphdegree_, j_, k_] :=
  (Delta[k - j] graphdegree^2 - Sum[G[[j, i]] G[[k, i]], {i, 1, N}]) / (N graphdegree^2);
DfDB[G_, N_, graphdegree_, j_, k_] := 0;
BstarDB = 1/N;
```

β

Numerical comparison for the population of size 12

```
βDBexemple = GetBeta[sBfDB, DfDB, G12generic,
   GE12generic, Q12generic, 12, 1, BstarDB /. N → 12] // FullSimplify
```

$$-\frac{1}{\mu}(-1+p)\,p\,\Big(\big(-1+dself^2\big)\big(eself+2\,ein\,Qin+9\,eout\,Qout\big)+2\,din^2\big(ein+eself+3\,ein\,Qin+eself\,Qin+18\,eout\,Qout\big)+18\,dout\,dself\big(eout+2\,eout\,Qin+2\,ein\,Qout+6\,eout\,Qout+eself\,Qout\big)+9\,dout^2\big((2\,ein+6\,eout+eself)(1+2\,Qin)+3\,(4\,ein+21\,eout+2\,eself)\,Qout\big)+4\,din\,\big(dself\,(ein+(ein+eself)\,Qin+9\,eout\,Qout)+9\,dout\,(eout+2\,eout\,Qin+2\,ein\,Qout+6\,eout\,Qout+eself\,Qout)\big)\Big)(-1+\mu)$$

```
βDBexemple - βDB /. {n → 3, d → 4} // FullSimplify
```

0

The difference is zero: we are fine!

Numerical comparison for the population of size 10

```
βDBexemple2 = GetBeta[sBfDB, DfDB, G10generic,
   GE10generic, Q10generic, 10, 1, BstarDB /. N → 10] // FullSimplify
```

$$-\frac{1}{\mu}(-1+p)\,p\,\Big(\big(-1+dself^2\big)\big(eself+4\,ein\,Qin+5\,eout\,Qout\big)+4\,din^2\big(eself+3\,eself\,Qin+ein\,(3+13\,Qin)+20\,eout\,Qout\big)+5\,dout^2\big((4\,ein+eself)(1+4\,Qin)+25\,eout\,Qout\big)+10\,dout\,dself\big(eout+4\,eout\,Qin+4\,ein\,Qout+eself\,Qout\big)+8\,din\,\big(dself\,(ein+3\,ein\,Qin+eself\,Qin+5\,eout\,Qout)+5\,dout\,(eout+4\,eout\,Qin+4\,ein\,Qout+eself\,Qout)\big)\Big)(-1+\mu)$$

```
βDBexemple2 - βDB /. {n → 5, d → 2} // FullSimplify
```

0

The difference is zero: we are fine!

γ

Numerical comparison for the population of size 12

```
γDBexemple = GetGamma[sBfDB, DfDB, G12generic,
   GE12generic, Q12generic, 12, 1, BstarDB /. N → 12] // FullSimplify
```

$$-\frac{1}{\mu}(-1+p)\,p\,\big(-1+dself^2+2\,din^2\,(1+Qin)+4\,din\,(dself\,Qin+9\,dout\,Qout)+9\,dout\,(dout+2\,dout\,Qin+6\,dout\,Qout+2\,dself\,Qout)\big)(-1+\mu)$$

```
γDBexemple - γDB /. {n → 3, d → 4} // FullSimplify
```

0

The difference is zero: we are fine!

Numerical comparison for the population of size 10

```
γDBexemple2 = GetGamma[sBfDB, DfDB, G10generic,
   GE10generic, Q10generic, 10, 1, BstarDB /. N → 10] // FullSimplify
```

$$-\frac{1}{\mu}(-1+p)\,p\,\left(-1+\text{dself}^2+4\,\text{din}^2\,(1+3\,\text{Qin})+8\,\text{din}\,(\text{dself Qin}+5\,\text{dout Qout})+5\,\text{dout}\,(\text{dout}+4\,\text{dout Qin}+2\,\text{dself Qout})\right)(-1+\mu)$$

```
γDBexemple2 - γDB /. {n → 5, d → 2} // FullSimplify
```

0

The difference is zero: we are fine!

Moran BD

The structure is the same as for Moran DB above - comments are lighter here!

Define δB and δD

$$\text{sBfBD[G\_, N\_, graphdegree\_, j\_, k\_]} := \frac{\text{Delta}[k-j]\,N-1}{N^2};$$

$$\text{DfBD[G\_, N\_, graphdegree\_, j\_, k\_]} := \frac{G[[k,j]]}{N\,\text{graphdegree}}-\frac{1}{N^2};$$

$$\text{BstarBD} = \frac{1}{N};$$

Equations β

```
βBDexemple = GetBeta[sBfBD, DfBD, G12generic,
   GE12generic, Q12generic, 12, 1, BstarBD /. N → 12] // FullSimplify
```

$$\frac{1}{12\,\mu}(-1+p)\,p\,\Big(12\,\big((-1+\text{dself})\,(\text{eself}+2\,\text{ein Qin}+9\,\text{eout Qout})+2\,\text{din}\,(\text{ein}+(\text{ein}+\text{eself})\,\text{Qin}+9\,\text{eout Qout})+9\,\text{dout}\,(\text{eout}+2\,\text{eout Qin}+2\,\text{ein Qout}+6\,\text{eout Qout}+\text{eself Qout})\big)+\big(-9\,\text{eout}+11\,\text{eself}-2\,(\text{ein}-10\,\text{ein Qin}+(9\,\text{eout}+\text{eself})\,\text{Qin})-9\,(2\,\text{ein}-3\,\text{eout}+\text{eself})\,\text{Qout}\big)\,\mu\Big)$$

Check also the different terms separately

$$\beta\text{IBDtest} = \text{Sum}\left[\left(\frac{\mu}{N^2}-\frac{G[[l,j]]}{N}\right)\text{GE}[[k,l]]\,Q[[j,k]],\{j,1,N\},\{k,1,N\},\{l,1,N\}\right] /.$$

```
    {G → G12generic, GE → GE12generic, Q → Q12generic, N → 12} //
   FullSimplify // Quiet
```

$$-\text{dself}\,(\text{eself}+2\,\text{ein Qin}+9\,\text{eout Qout})-2\,\text{din}\,(\text{ein}+(\text{ein}+\text{eself})\,\text{Qin}+9\,\text{eout Qout})-9\,\text{dout}\,(\text{eout}+2\,\text{eout Qin}+2\,\text{ein Qout}+6\,\text{eout Qout}+\text{eself Qout})+\frac{1}{12}(2\,\text{ein}+9\,\text{eout}+\text{eself})\,(1+2\,\text{Qin}+9\,\text{Qout})\,\mu$$

```
βIBDtest + βBDI /. {n → 3, d → 4} // FullSimplify
```

0

```
βDBDtest = Sum[(1 - μ)/N GE[[k, l]] Q[[l, k]], {k, 1, N}, {l, 1, N}] /. {G → G12generic,
    GE → GE12generic, Q → Q12generic, N → 12} // FullSimplify // Quiet
```

$$-(\text{eself} + 2\,\text{ein}\,\text{Qin} + 9\,\text{eout}\,\text{Qout})\,(-1+\mu)$$

```
βDBDtest - βBDD /. {n → 3, d → 4} // FullSimplify
```

0

```
βBDtest = p (1 - p)/μ (βDBDtest + βIBDtest) // FullSimplify;
```

Numerical comparison (the differences are zero: we are fine!)

```
βBDtest - βBD /. {n → 3, d → 4} // FullSimplify
```

0

The difference is zero: we are fine!

```
βBDD - βDBDtest /. {n → 3, d → 4} // FullSimplify
```

0

The difference is zero: we are fine!

```
βBDI + βIBDtest /. {n → 3, d → 4} // FullSimplify
```

0

The difference is zero: we are fine!

Equations γ

```
γBDexemple = GetGamma[sBfBD, DfBD, G12generic,
   GE12generic, Q12generic, 12, 1, BstarBD /. N → 12] // FullSimplify
```

$$\frac{(-1+p)\,p\,\left(12\,(-1+\text{dself}+2\,\text{din}\,\text{Qin}+9\,\text{dout}\,\text{Qout})+(11-2\,\text{Qin}-9\,\text{Qout})\,\mu\right)}{12\,\mu}$$

```
γBDexemple - γBD /. {n → 3, d → 4} // Simplify
```

0

The difference is zero: we are fine!

Wright-Fisher

The structure is the same as for Moran DB above - comments are lighter here!

Define δB and δD

```
sBfWF[G_, N_, graphdegree_, j_, k_] :=
  1/graphdegree^2 (Delta[k - j] graphdegree^2 - Sum[G[[j, i]] G[[k, i]], {i, 1, N}]);
DfWF[G_, N_, graphdegree_, j_, k_] := 0;
BstarWF = 1;
```

Equations β

```
βWFexemple = GetBeta[sBfWF, DfWF, G12generic,
   GE12generic, Q12generic, 12, 1, BstarWF /. N → 12] // FullSimplify
```

$$-\frac{1}{\mu}(-1+p)\,p\,\Big((-1+\text{dself}^2)\,(\text{eself}+2\,\text{ein}\,\text{Qin}+9\,\text{eout}\,\text{Qout})+ 2\,\text{din}^2\,(\text{ein}+\text{eself}+3\,\text{ein}\,\text{Qin}+\text{eself}\,\text{Qin}+18\,\text{eout}\,\text{Qout})+ 18\,\text{dout}\,\text{dself}\,(\text{eout}+2\,\text{eout}\,\text{Qin}+2\,\text{ein}\,\text{Qout}+6\,\text{eout}\,\text{Qout}+\text{eself}\,\text{Qout})+ 9\,\text{dout}^2\,\big((2\,\text{ein}+6\,\text{eout}+\text{eself})\,(1+2\,\text{Qin})+3\,(4\,\text{ein}+21\,\text{eout}+2\,\text{eself})\,\text{Qout}\big)+ 4\,\text{din}\,\big(\text{dself}\,(\text{ein}+(\text{ein}+\text{eself})\,\text{Qin}+9\,\text{eout}\,\text{Qout})+ 9\,\text{dout}\,(\text{eout}+2\,\text{eout}\,\text{Qin}+2\,\text{ein}\,\text{Qout}+6\,\text{eout}\,\text{Qout}+\text{eself}\,\text{Qout})\big)\Big)\,(-1+\mu)$$

```
βWFexemple - βWF /. {n → 3, d → 4} // Simplify
```

0

The difference is zero: we are fine!

Equations γ

```
γWFexemple = GetGamma[sBfWF, DfWF, G12generic,
   GE12generic, Q12generic, 12, 1, BstarWF /. N → 12] // FullSimplify
```

$$-\frac{1}{\mu}(-1+p)\,p\,\Big(-1+\text{dself}^2+2\,\text{din}^2\,(1+\text{Qin})+4\,\text{din}\,(\text{dself}\,\text{Qin}+9\,\text{dout}\,\text{Qout})+ 9\,\text{dout}\,(\text{dout}+2\,\text{dout}\,\text{Qin}+6\,\text{dout}\,\text{Qout}+2\,\text{dself}\,\text{Qout})\Big)\,(-1+\mu)$$

```
γWFexemple - γWF /. {n → 3, d → 4} // Simplify
```

0

The difference is zero: we are fine!

Expected frequencies of altruists in the population

Moran BD

```
EXBD = p + δ (βBD b - γBD c) /. {Qin → QinM, Qout → QoutM} /. genericde // Simplify
```

```
(p ((b - c) (-1 + p) δ μ (m - n + Idself (-1 + m) (-1 + μ) + μ - m μ) +
    d^2 (-c Idself m n δ - c m^2 n δ + c Idself m^2 n δ + c m n^2 δ + c Idself m n p δ + c m^2 n p δ -
       c Idself m^2 n p δ - c m n^2 p δ + Idself μ - Idself m μ - n μ + 2 m n μ - m n^2 μ +
       2 c Idself δ μ - 2 c Idself m δ μ - 2 c n δ μ - c Idself n δ μ + 2 c m n δ μ +
       2 c Idself m n δ μ + c m^2 n δ μ - c Idself m^2 n δ μ + c n^2 δ μ - 2 c m n^2 δ μ -
       2 c Idself p δ μ + 2 c Idself m p δ μ + 2 c n p δ μ + c Idself n p δ μ - 2 c m n p δ μ -
       2 c Idself m n p δ μ - c m^2 n p δ μ + c Idself m^2 n p δ μ - c n^2 p δ μ + 2 c m n^2 p δ μ -
       Idself μ^2 + Idself m μ^2 + 2 n μ^2 - 2 m n μ^2 - n^2 μ^2 + m n^2 μ^2 - c Idself δ μ^2 +
       c Idself m δ μ^2 + 2 c n δ μ^2 - 2 c m n δ μ^2 - c n^2 δ μ^2 + c m n^2 δ μ^2 + c Idself p δ μ^2 -
       c Idself m p δ μ^2 - 2 c n p δ μ^2 + 2 c m n p δ μ^2 + c n^2 p δ μ^2 - c m n^2 p δ μ^2 +
       b (-1 + g) (-1 + p) δ (-Ieself (m (-1 + μ) - μ) (m + n (-1 + μ) - μ) + (-1 + m)
             n (-2 + μ) μ + Idself (-1 + m) (Ieself (m (-1 + μ) - μ) - (-2 + μ) μ))) +
    d (m^2 (-1 + μ) (-1 + c (-1 + p) δ + b (δ - p δ) + μ) + μ ((c + b (-1 + (-1 + g) Ieself))
             (-1 + p) δ μ + n (1 + 3 c δ - 3 c p δ - 2 μ - 2 c δ μ + 2 c p δ μ -
                b (-1 + p) δ (-3 + Ieself + g (2 + Ieself (-1 + μ) - μ) + μ - Ieself μ)) +
          n^2 (μ + c δ (-1 + p + μ - p μ))) + m (n (b (δ - p δ) - (-1 + c (-1 + p) δ) (-1 + μ)) -
          (-1 + p) δ μ (2 c (-1 + μ) + b (2 - Ieself - 2 μ + g (-2 + Ieself + μ)))) + Idself
       (-1 + m) (m (1 + b (-1 + p) δ + c (δ - p δ) - μ) (-1 + μ) + μ (1 - μ + c (-1 + p) δ
             (-3 + n + 2 μ) + b (-1 + p) δ (3 - Ieself - 2 μ + g (-2 + Ieself + μ))))))) /
 (d (m^2 (-1 + μ)^2 - Idself (-1 + m) (-1 + μ) (m (-1 + μ) + μ - d μ) -
     (-1 + d) n μ (1 + (-2 + n) μ) +
     m n (-1 + μ) (1 + d (-2 + n) μ)))
```

Moran DB

```
EXDB = p + δ (βDB b - γDB c) /. {Qin → QinM, Qout → QoutM} /. genericde // Simplify
```

```
(p ((-1 + n) (n - d n)^3 (-m^2 (-1 + μ)^2 + Idself (-1 + m) (-1 + μ) (m (-1 + μ) + μ - d μ) +
      (-1 + d) n μ (1 + (-2 + n) μ) - m n (-1 + μ) (1 + d (-2 + n) μ)) + (-1 + d)^2 n^3 (1 - p) δ
   (-1 + μ) (c (-1 + d) (-1 + n) (m n (2 + d (n (-1 + μ) - 2 μ)) - (-1 + d) (-2 + n) n μ +
        m^2 (-2 + d n + μ) - Idself^2 (-1 + m)^2 (d (m (-1 + μ) - μ) + μ) +
        Idself (-1 + m) (d m^2 (-1 + μ) + 2 (-1 + d) μ + m (2 - 2 d μ))) +
     b (2 m^2 - 2 d m^2 - 2 d Ieself m^2 + 2 d^2 Ieself m^2 + 2 d g Ieself m^2 - 2 d^2 g Ieself m^2 +
        d g m^3 + d Ieself m^3 - d^2 Ieself m^3 - d g Ieself m^3 + d^2 g Ieself m^3 - 2 m n +
        2 d m n + 2 d Ieself m n - 2 d^2 Ieself m n - 2 d g Ieself m n + 2 d^2 g Ieself m n -
        2 m^2 n + 3 d m^2 n - d^2 m^2 n - 2 d g m^2 n + d^2 g m^2 n - d m^3 n + d^2 m^3 n - d^2 g m^3 n +
        2 m n^2 - 3 d m n^2 + d^2 m n^2 + d g m n^2 - d^2 g m n^2 - d Ieself m n^2 + d^2 Ieself m n^2 +
        d g Ieself m n^2 - d^2 g Ieself m n^2 + d m^2 n^2 - d^2 m^2 n^2 + d^2 g m^2 n^2 + 2 g m μ -
        2 d g m μ + 2 Ieself m μ - 4 d Ieself m μ + 2 d^2 Ieself m μ - 2 g Ieself m μ +
        4 d g Ieself m μ - 2 d^2 g Ieself m μ - m^2 μ + d m^2 μ - g m^2 μ + 2 d g m^2 μ -
        Ieself m^2 μ + 4 d Ieself m^2 μ - 3 d^2 Ieself m^2 μ + g Ieself m^2 μ -
        4 d g Ieself m^2 μ + 3 d^2 g Ieself m^2 μ - d g m^3 μ - d Ieself m^3 μ + d^2 Ieself m^3 μ +
        d g Ieself m^3 μ - d^2 g Ieself m^3 μ + 2 n μ - 4 d n μ + 2 d^2 n μ - 2 g n μ + 4 d g n μ -
        2 d^2 g n μ - 2 Ieself n μ + 4 d Ieself n μ - 2 d^2 Ieself n μ + 2 g Ieself n μ -
        4 d g Ieself n μ + 2 d^2 g Ieself n μ - 2 m n μ + 6 d m n μ - 4 d^2 m n μ - 4 d g m n μ +
        4 d^2 g m n μ - 2 d Ieself m n μ + 2 d^2 Ieself m n μ + 2 d g Ieself m n μ -
        2 d^2 g Ieself m n μ + 2 m^2 n μ - 5 d m^2 n μ + 3 d^2 m^2 n μ + 2 d g m^2 n μ - 3 d^2 g m^2 n μ +
        d m^3 n μ - d^2 m^3 n μ + d^2 g m^3 n μ - n^2 μ + 2 d n^2 μ - d^2 n^2 μ + g n^2 μ - 2 d g n^2 μ +
        d^2 g n^2 μ + Ieself n^2 μ - 2 d Ieself n^2 μ + d^2 Ieself n^2 μ - g Ieself n^2 μ +
        2 d g Ieself n^2 μ - d^2 g Ieself n^2 μ - d m n^2 μ + d^2 m n^2 μ + d g m n^2 μ - d^2 g m n^2 μ +
        d Ieself m n^2 μ - d^2 Ieself m n^2 μ - d g Ieself m n^2 μ + d^2 g Ieself m n^2 μ -
        (-1 + d) (-1 + g) Idself^2 (-1 + Ieself) (-1 + m)^2 (d (m (-1 + μ) - μ) + μ) +
        Idself (-1 + m) (d m^2 (-1 - 2 g (-1 + Ieself) + 2 Ieself +
              d (-1 + g) (-1 + 2 Ieself - n) - n) (-1 + μ) + 2 (-1 + d)^2 (-1 + g)
            (-1 + Ieself) μ - 2 (-1 + d) m (-1 + n + g μ + Ieself μ - g Ieself μ -
              n μ - d (-1 + g) (Ieself + n (-1 + μ) + μ - 2 Ieself μ)))))) /
 ((-1 + n) (n - d n)^3 (-m^2 (-1 + μ)^2 + Idself (-1 + m) (-1 + μ)
     (m (-1 + μ) + μ - d μ) + (-1 + d)
     n
     μ
     (1 + (-2 + n) μ) - m
     n
     (-1 + μ)
     (1 + d (-2 + n) μ)))
```

Wright - Fisher

```
EXWF = p + δ (βWF b - γWF c) /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify
```

$$p+$$

$$\frac{1}{\mu}(1-p)\,p\,\delta\left(-c\left(1-\mu-(1-\mu)\left(\frac{\text{Idself}^2(-1+m)^2}{n^2}+\frac{(-1+m)^2(\text{Idself}-n)^2}{(-1+n)\,n^2}+\frac{m^2}{(-1+d)\,n}-\right.\right.\right.$$

$$\frac{(2+d(-2+m))\,m\left(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\right)}{(-1+d)\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}+$$

$$\Bigg(\left(2(-1+d)\,\text{Idself}\,(-1+m)^2-(-1+d)\,\text{Idself}^2(-1+m)^2+\right.$$

$$\left.(-1+d)(-2+n)\,n-2(-1+d)\,m(-2+n)\,n+m^2(1+d(-2+n)\,n)\right)$$

$$\left.\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\right)\Bigg/$$

$$\left((-1+d)(-1+n)\,n\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\right.\right.$$

$$\left.\left.\left.\left.\left.\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\right)\right)\right)+$$

$$b(1-\mu)\left(\frac{\text{Ieself}-g\,\text{Ieself}}{n}+\frac{g\left(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Ieself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}}-\right.$$

$$\frac{(1-g)\left(\text{Ieself}-n\right)\left(\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{n\left(\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}-$$

$$\frac{1}{n^2}\left(\text{Idself}-\text{Idself}\,m\right)^2\left(\frac{\text{Ieself}-g\,\text{Ieself}}{n}+\frac{g\left(-\frac{1}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}-\right.$$

$$\left.\frac{(1-g)\left(\text{Ieself}-n\right)\left(\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{n\left(\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}\right)-$$

$$\frac{1}{(-1+d)^2\,n}\,m^2\left(\frac{\left(2-3\,d+d^2+g\right)n\left(-\frac{1}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}+\right.$$

$$\left.\left(-1+d-g\right)\left(1+\frac{(-1+n)\left(\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}\right)\right)-$$

$$\frac{1}{n}\,2\,\text{Idself}\,(1-m)\,m\left(\frac{(1-g)\left(-\frac{1}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}+\right.$$

$$\frac{1}{(-1+d)\,n}\,g\left(1+\frac{(-2+d)\,n\left(-\frac{1}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{\left(1+d\,(-1+m)\right)^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}+\right.$$

$$\frac{(-1+n)\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}}\Bigg)\Bigg)-$$

$$\frac{1}{(-1+n)\,n^2}(-1+m)^2(\text{Idself}-n)^2\Bigg(\frac{\text{Ieself}-g\,\text{Ieself}}{n}+$$

$$\frac{g(-1+n)\left(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}}+$$

$$\frac{(1-g)\,\text{Ieself}\,(-2+n)\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}{n\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}+$$

$$\frac{(1-g)(\text{Ieself}-n)\left(-2+n+\frac{(3+(-3+n)n)\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}}\right)}{n-n^2}\Bigg)+$$

$$\frac{1}{n}2(1-m)(\text{Idself}-n)\Bigg(\frac{1}{-1+d}m\Bigg(\frac{g}{n}+$$

$$\frac{(-1+d-g)\left(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}}+$$

$$\frac{g(-1+n)\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}{n\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}\Bigg)+\frac{1}{n}\mathtt{Idself}\,(1-$$

$$m)\Bigg(\frac{(-1+g)(\mathtt{Ieself}-n)}{(-1+n)\,n}+\frac{g\left(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\right)}{\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}}+$$

$$\frac{(1-g)\,\mathtt{Ieself}\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}{n\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)}+$$

$$\Bigg((1-g)(\mathtt{Ieself}-n)(-2+n)\Bigg(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-$$

$$\frac{d}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Big/\Bigg((n-n^2)$$

$$\Bigg(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathtt{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg)\Bigg)\Bigg)$$

Export to R

Export the EX formulas to R

Rewrite the Greek letters

```
GreekTerms = {δ → sel, μ → mut}
```

```
{δ → sel, μ → mut}
```

Common parts to all functions

```
FunctionPartB = " <- function(b, c, p, sel, mut, m, g, n, d, Idself, Ieself){
## Arguments:
#  b   benefit of interaction
#  c   cost of interaction
#  p   mutation bias
#  sel intensity of selection
#  mut mutation probability
#  m   emigration probability
#  g   proportion of interactions
     out of the group (interaction equivalent of m)
#  n   deme size
#  d   number of demes
#  Idself whether reproduction in site where the parent is
#  Ieself whether interactions with oneself
return(
";
FunctionPartE = ")
}";
```

Function to translate Mathematica to R

```
ToRForm[x_] := ToString[x /. GreekTerms // CForm]
```

Do it for all life cycles

```
RtxtBD = "pBD " <> FunctionPartB <> ToRForm[EXBD] <> FunctionPartE;
RtxtDB = "pDB " <> FunctionPartB <> ToRForm[EXDB] <> FunctionPartE;
RtxtWF = "pWF " <> FunctionPartB <> ToRForm[EXWF] <> FunctionPartE;
```

Define Power function in R

```
PowerDef = "Power <- function(a,b) return(a^b)";
```

Combine all texts

```
Rtxt = PowerDef <> "

" <> RtxtBD <> "

" <> RtxtDB <> "

" <> RtxtWF;
```

Export to txt file (Mathematica did not want R)

```
Export[pathtosave <> "Mathematica/analytics.txt", Rtxt];
```

Convert the file extension to R

```
cmd = "mv" <> " " <> pathtosave <>
    "Mathematica/analytics.txt " <> pathtosave <> "Mathematica/analytics.R";
```

```
Get["!" <> cmd];
```

Export to R the β and γ functions

Rewrite the Greek letters

```
GreekTerms = {ω → sel, μ → mut}
```

```
{ω → sel, μ → mut}
```

Common parts to all functions

```
FunctionPartB = " <- function(p, sel, mut, m, g, n, d, Idself, Ieself){
## Arguments:
#  p   mutation bias
#  sel intensity of selection
#  mut mutation probability
#  m   emigration probability
#  g   proportion of interactions
     out of the group (interaction equivalent of m)
#  n   deme size
#  d   number of demes
#  Idself whether reproduction in site where the parent is
#  Ieself whether interactions with oneself
return(
";
FunctionPartE = ")
}";
```

Function to translate Mathematica to R

```
ToRForm[x_] := ToString[x /. GreekTerms // CForm]
```

Do it for β and γ

```
RtxtbBDD = "bBDD " <> FunctionPartB <>
    ToRForm[βBDD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
RtxtbBDI = "bBDI " <> FunctionPartB <>
    ToRForm[βBDI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
RtxtcBDD = "cBDD " <> FunctionPartB <>
    ToRForm[γBDD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
RtxtcBDI = "cBDI " <> FunctionPartB <>
    ToRForm[γBDI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
```

```
RtxtbDBD = "bDBD " <> FunctionPartB <>
    ToRForm[βDBD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
RtxtbDBI = "bDBI " <> FunctionPartB <>
    ToRForm[βDBI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
RtxtcDBD = "cDBD " <> FunctionPartB <>
    ToRForm[γDBD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
RtxtcDBI = "cDBI " <> FunctionPartB <>
    ToRForm[γDBI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <>
    FunctionPartE;
```

```
RtxtbWFD = "bWFD " <> FunctionPartB <> ToRForm[
     βWFD /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
RtxtbWFI = "bWFI " <> FunctionPartB <> ToRForm[
     βWFI /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
RtxtcWFD = "cWFD " <> FunctionPartB <> ToRForm[
     γWFD /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
```

```
RtxtcWFI = "cWFI " <> FunctionPartB <> ToRForm[
     γWFI /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
```

Define Power function in R

```
PowerDef = "Power <- function(a,b) return(a^b)";
```

Combine all texts

```
Rtxt = PowerDef <> "

" <> RtxtbBDD <> "

" <> RtxtbBDI <> "

" <> RtxtcBDD <> "

" <> RtxtcBDI <> "

" <> RtxtbDBD <> "

" <> RtxtbDBI <> "

" <> RtxtcDBD <> "

" <> RtxtcDBI <> "

" <> RtxtbWFD <> "

" <> RtxtbWFI <> "

" <> RtxtcWFD <> "

" <> RtxtcWFI;
```

Export to txt file (Mathematica did not want R)

```
Export[pathtosave <> "Mathematica/analyticsBC.txt", Rtxt];
```

Convert the file extension to R

```
cmd = "mv " <> pathtosave <> "Mathematica/analyticsBC.txt " <>
  pathtosave <> "Mathematica/analyticsBC.R"

Get["!" <> cmd];
```