**Supplementary Mathematica file.**
July 2017

**CONTENTS:**

**Part 0**: “Housekeeping”, definitions of functions, matrices and replacement rules that will be used throughout the file.

**Part 1**: Probabilities of identity by descent (related to Appendix B2 and C2).
We compare the formulas calculated by hand to the ones obtained numerically for special structures, to check that the formulas are correct (they are).

**Part 2**: Expected Frequencies Functions (related to Appendix B1 and C1, and formulas in the main text)
- We simplify the formulas obtained by hand by replacing the dispersal (d) and interaction (e) graphs by their formulas in subdivided populations.
- We compare the formulas calculated by hand to the ones obtained numerically for special structures, to check that the formulas are correct (they are).
- We export the formulas to R for further use.

**Part 3**: Changes with *m*
We study how the different terms of our equations (probabilities of identity by descent Q, indirect/secondary terms I, and full equations EX) change with the emigration probability *m*, and identify critical values.

**Please note:**
- In this file, the mutation bias is denoted by p (as in Tarnita and Taylor 2014 and Debarre 2017), instead of *v* as in the manuscript.
The letter *v* was chosen in the manuscript because p was sometimes mistaken by others as average frequency of altruists in the population ($\overline{X}$ in the manuscript)... But this file was written before the change, and it is too complicated to change every instance of “p”.

- Make sure to change `pathtosave` with the path to the folder containing the codes.

Before doing anything, clean the memory

```
In[1]:= Clear[Evaluate[Context[] <> "*"]]
```

Set path to folder where outputs should be saved (otherwise it is the default Mathematica one)

```
In[2]:= pathtosave = "~/Documents/Work/Projects/2016_SocEvolSubdivPop/Programs/";
```

# 0) Generalities - Initializations

# Some functions

Function to turn P (expected state of pairs of sites) into Q (probabilities of identity by descent)

In[3]:= 
```
PtoQ[P_] := (P - p^2)/(p (1 - p)) // FullSimplify;
```

$$\text{PtoQ[P\_]} := \frac{P - p^2}{p\,(1 - p)} \text{ // FullSimplify;}$$

Delta function

In[4]:= 
```
Delta[x_] := If[x == 0, 1, 0]
```

# Define graphs for numerical evaluation

## Dispersal and Interaction Graphs

### Island model, dispersal graph, generic

N = 12, 4 demes of 3 individuals

In[5]:= 
```
G12generic =
```

$$\begin{pmatrix}
dself & din & din & dout & dout & dout & dout & dout & dout & dout & dout & dout \\
din & dself & din & dout & dout & dout & dout & dout & dout & dout & dout & dout \\
din & din & dself & dout & dout & dout & dout & dout & dout & dout & dout & dout \\
dout & dout & dout & dself & din & din & dout & dout & dout & dout & dout & dout \\
dout & dout & dout & din & dself & din & dout & dout & dout & dout & dout & dout \\
dout & dout & dout & din & din & dself & dout & dout & dout & dout & dout & dout \\
dout & dout & dout & dout & dout & dout & dself & din & din & dout & dout & dout \\
dout & dout & dout & dout & dout & dout & din & dself & din & dout & dout & dout \\
dout & dout & dout & dout & dout & dout & din & din & dself & dout & dout & dout \\
dout & dout & dout & dout & dout & dout & dout & dout & dout & dself & din & din \\
dout & dout & dout & dout & dout & dout & dout & dout & dout & din & dself & din \\
dout & dout & dout & dout & dout & dout & dout & dout & dout & din & din & dself
\end{pmatrix};$$

```
Nin = 3;
Ndemes = 4;
```

N = 10, 2 demes of 5 individuals

In[8]:= 
$$\text{G10generic} = \begin{pmatrix}
dself & din & din & din & din & dout & dout & dout & dout & dout \\
din & dself & din & din & din & dout & dout & dout & dout & dout \\
din & din & dself & din & din & dout & dout & dout & dout & dout \\
din & din & din & dself & din & dout & dout & dout & dout & dout \\
din & din & din & din & dself & dout & dout & dout & dout & dout \\
dout & dout & dout & dout & dout & dself & din & din & din & din \\
dout & dout & dout & dout & dout & din & dself & din & din & din \\
dout & dout & dout & dout & dout & din & din & dself & din & din \\
dout & dout & dout & dout & dout & din & din & din & dself & din \\
dout & dout & dout & dout & dout & din & din & din & din & dself
\end{pmatrix};$$

## Island model, interaction graph, generic

In[9]:=
```
GE12generic = G12generic /. {dself → eself, dout → eout, din → ein};
GE10generic = G10generic /. {dself → eself, dout → eout, din → ein};
```

## Formulas for d and e

Replacements for the generic dispersal probabilities, depending on whether there is self-replacement or not

In[11]:=
```
noselfreplacement = {dself → 0, din → (1 - m)/(n - 1), dout → m/(d n - n)};
withselfreplacement = {dself → (1 - m)/n, din → (1 - m)/n, dout → m/(d n - n)};
```

Replacements for the generic interaction probabilities, depending on whether there is self - interaction or not

In[13]:=
```
groupnoself = {eself → 0, ein → 1/(n - 1), eout → 0};
groupwithself = {eself → 1/n, ein → 1/n, eout → 0};
```

We can even assume that there are a proportion g of interactions outside of the group

In[15]:=
```
widewithself = {eself → (1 - g)/n, ein → (1 - g)/n, eout → g/(n d - n)};
widenoself = {eself → 0, ein → (1 - g)/(n - 1), eout → g/(n d - n)};
```

Combine these using Idself and Ieself, indicator variables for whether there is dispersal/interaction with self

In[17]:=
```
genericde = {dself → Idself (1 - m)/n, din → (1 - m) (1/(n - 1) - Idself 1/(n (n - 1))), dout → m/(n d - n), (*
   *)eself → Ieself (1 - g)/n, ein → (1 - g) (1/(n - 1) - Ieself 1/(n (n - 1))), eout → g/(n d - n)};
```

Quick check

In[18]:=
```
eself + (n - 1) ein + (n d - n) eout /. genericde // Simplify
dself + (n - 1) din + (n d - n) dout /. genericde // Simplify
```

Out[18]= 1

Out[19]= 1

# Probabilities of identity by descent matrices

## Generic Q matrix corresponding to the populations defined above

N = 12

In[20]:= 

$$\text{Q12generic} = \begin{pmatrix} 1 & \text{Qin} & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qin} & 1 & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qin} & \text{Qin} & 1 & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & 1 & \text{Qin} & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & 1 & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & \text{Qin} & 1 & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & 1 & \text{Qin} & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & 1 & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & \text{Qin} & 1 & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & 1 & \text{Qin} & \text{Qin} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & 1 & \text{Qin} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & \text{Qin} & 1 \end{pmatrix};$$

N = 10

In[21]:= 

$$\text{Q10generic} = \begin{pmatrix} 1 & \text{Qin} & \text{Qin} & \text{Qin} & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qin} & 1 & \text{Qin} & \text{Qin} & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qin} & \text{Qin} & 1 & \text{Qin} & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qin} & \text{Qin} & \text{Qin} & 1 & \text{Qin} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qin} & \text{Qin} & \text{Qin} & \text{Qin} & 1 & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & 1 & \text{Qin} & \text{Qin} & \text{Qin} & \text{Qin} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & 1 & \text{Qin} & \text{Qin} & \text{Qin} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & \text{Qin} & 1 & \text{Qin} & \text{Qin} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & \text{Qin} & \text{Qin} & 1 & \text{Qin} \\ \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qout} & \text{Qin} & \text{Qin} & \text{Qin} & \text{Qin} & 1 \end{pmatrix};$$

# 1) Probabilities of identity by descent (Q)

## Moran

### Simplify QinM and QoutM

See Appendix B2 for calculation details on how Qself, Qin and Qout were obtained using a formula presented in the appendix of Debarre 2017 JTB for “2D graphs”. Here we just copy these formulas.

In[22]:= 

$$\text{QselfM2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{\mu} + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(n-1) + \frac{1}{1-(1-\mu)\,\left(1-m-\frac{m}{d-1}\right)}\,(d-1) + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(d-1)\,(n-1)\right);$$

In[23]:=
$$\text{QinM2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{\mu} + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(-1) + \frac{1}{1-(1-\mu)\,\left(1-m-\frac{m}{d-1}\right)}\,(d-1) + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(d-1)\,(-1)\right);$$

$$\text{QoutM2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{\mu} + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\,(-1) + \frac{1}{1-(1-\mu)\,\left(1-m-\frac{m}{d-1}\right)}\,(-1) + \frac{1}{1-(1-\mu)\,(\text{dself}-\text{din})}\right);$$

Find $\lambda$ using Qself == 1

In[25]:=
```
theλM = λ /. Solve[QselfM2 == 1, λ][[1]] // FullSimplify
```

Out[25]=
$$\left(n\,(1+\text{din}+\text{dself}\,(-1+\mu)-\text{din}\,\mu)\,(-d\,m+\mu+d\,(-1+m)\,\mu)\right) \Big/ \left(m\,(-1+\mu)\,\left(1+\text{din}-\text{dself}+(-\text{din}+\text{dself}+d\,(-1+n))\,\mu\right) + (-1+d)\,\mu\,\left(-1+\text{dself}+\text{din}\,(-1+\mu)+\mu-(\text{dself}+n)\,\mu\right)\right)$$

Replace $\lambda$ in the equations for Qin and Qout

In[26]:=
```
QinM = QinM2 /. λ → theλM // FullSimplify
QoutM = QoutM2 /. λ → theλM // FullSimplify
```

Out[26]=
$$\left((-1+\mu)\,\left((-1+d)\,(1+\text{din}-\text{dself})\,\mu+m\,\left(1+\text{din}-\text{dself}-(d+\text{din}-\text{dself})\,\mu\right)\right)\right) \Big/ \left(m\,(-1+\mu)\,\left(1+\text{din}-\text{dself}+(-\text{din}+\text{dself}+d\,(-1+n))\,\mu\right) + (-1+d)\,\mu\,\left(-1+\text{dself}+\text{din}\,(-1+\mu)+\mu-(\text{dself}+n)\,\mu\right)\right)$$

Out[27]=
$$\left(m\,(-1+\mu)\,(1+\text{din}+\text{dself}\,(-1+\mu)-\text{din}\,\mu)\right) \Big/ \left(m\,(-1+\mu)\,\left(1+\text{din}-\text{dself}+(-\text{din}+\text{dself}+d\,(-1+n))\,\mu\right) + (-1+d)\,\mu\,\left(-1+\text{dself}+\text{din}\,(-1+\mu)+\mu-(\text{dself}+n)\,\mu\right)\right)$$

## Check numerically

Here we evaluate the probabilities of identity by descent numerically, using the recursion formula (“eqs” in the function below), with specific graphs.

```
In[28]:= NGetQM[G_, N_, graphdegree_, p_, μ_] := Module[{QT, eqs, vars, sols, QTs}, (*
   G is the dispersal graph,
   N is the size of the population,
   graphdegree is the degree of the graph (=1 in a subdivided population),
   p is the mutation biais,
   μ is the mutation probability.
   *)

   (* Initialize the QT matrix *)
   Do[Q_{i,j} = 0; Q_{i,j} =., {i, 1, N}, {j, 1, N}];
   QT = Table[Q_{i,j}, {i, 1, N}, {j, 1, N}];
   Do[QT[[i, i]] = 1, {i, 1, N}]; (* Q_{i,i}= 1 *)
   Do[QT[[i, j]] = Q_{j,i}, {j, 1, N - 1}, {i, j + 1, N}]; (* Because Q is symmetric *)

   eqs =
    Flatten[Table[Q_{i,j} == (1 - μ)/(2 graphdegree) (Sum[G[[l, j]] QT[[l, i]] + G[[l, i]] QT[[l, j]], {l, 1, N}]),
      {i, 1, N - 1}, {j, i + 1, N}]];

   vars = Flatten[Table[Q_{i,j}, {i, 1, N - 1}, {j, i + 1, N}]];
   sols = NSolve[eqs, vars];

   QTs = QT /. sols[[1]];
   QTs]
```

This function compares the numerical version to the analytical one, for specific graph structures. If the numerical values are the same, we are fine! (and we are, otherwise there would not be a paper)

```
In[29]:= prs =.;
CheckQM[prs_, dvalues_] := Module[{NG, NQin, NQout, NQinMatrix},
   NG = ToExpression["G" <> ToString[d n /. prs] <> "generic"] /. dvalues /. prs;
   NQinMatrix = NGetQM[NG, n d /. prs, 1, 0.5, μ /. prs];
   NQin = NQinMatrix[[1, 2]];
   NQout = NQinMatrix[[1, n d /. prs]];
   Print[{{"Qin", NQin, QinM /. dvalues /. prs},
      {"Qout", NQout, QoutM /. dvalues /. prs}} // Transpose // MatrixForm]
  ]
```

Check for the population of size 12

```
In[31]:= CheckQM[{m → 0.2, d → 4, n → 3, μ → 0.2}, withselfreplacement]
CheckQM[{m → 0.2, d → 4, n → 3, μ → 0.2}, noselfreplacement]
```

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.407643 & 0.127389 \\ 0.407643 & 0.127389 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.503735 & 0.140875 \\ 0.503735 & 0.140875 \end{pmatrix}$$

Check for the population of size 10

In[33]:=
```
CheckQM[{m → 0.2, d → 2, n → 5, μ → 0.2}, withselfreplacement]
CheckQM[{m → 0.2, d → 2, n → 5, μ → 0.2}, noselfreplacement]
```

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.329897 & 0.206186 \\ 0.329897 & 0.206186 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.3762 & 0.222649 \\ 0.3762 & 0.222649 \end{pmatrix}$$

## Particular cases

### Equations with self replacement (dself = din)

In[35]:=
```
QinMs = QinM /. withselfreplacement // FullSimplify
QoutMs = QoutM /. withselfreplacement // FullSimplify
```

Out[35]= $-\frac{(-1+\mu)\left(\mu - d\,\mu + m\left(-1+d\,\mu\right)\right)}{-\left(-1+d\right)\mu\,(1+(-1+n)\,\mu)+m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)}$

Out[36]= $\frac{m\,(-1+\mu)}{-\left(-1+d\right)\mu\,(1+(-1+n)\,\mu)+m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)}$

Simplify the way the equations are written (human - friendly versions), and check that the formulas remain correct

In[37]:=
```
QinMs - ((1 - μ) (d μ (1 - m) - μ + m)) / ((d - 1) μ (1 + (n - 1) μ) + m (1 - μ) (1 + d (n - 1) μ)) //
 FullSimplify
```

Out[37]= 0

In[38]:=
```
QoutMs - (m (1 - μ)) / ((d - 1) μ (1 + (n - 1) μ) + m (1 - μ) (1 + d (n - 1) μ)) // FullSimplify
```

Out[38]= 0

Check limit behavior:

First infinite population, then zero mutation
vs.
First zero mutation, then infinite population

In[39]:=
```
Limit[QoutMs, d → ∞] // FullSimplify
Limit[Limit[QoutMs, μ → 0], d → ∞] // FullSimplify
```

Out[39]= 0

Out[40]= 1

In[41]:=
```
Limit[QinMs, d → ∞] // FullSimplify
% /. μ → 0 // FullSimplify
```

Out[41]= $$\frac{-1+m+\mu-m\mu}{-1+m(-1+n)(-1+\mu)+\mu-n\mu}$$

Out[42]= $$\frac{1-m}{1+m(-1+n)}$$

In[43]:=
```
Limit[QinMs, μ → 0]
Limit[%, d → ∞] // FullSimplify
```

Out[43]= 1

Out[44]= 1

In[45]:=
```
Series[QinMs, {μ, 0, 1}]
```

Out[45]= $1-d\,n\,\mu+O[\mu]^2$

## Equations without Self - replacement (dself = 0)

In[46]:=
```
QinMw = QinM /. noselfreplacement // FullSimplify
QoutMw = QoutM /. noselfreplacement // FullSimplify
```

Out[46]= $$\left((-1+\mu)\left(m^2(-1+\mu)+(-1+d)\,n\,\mu+m\,(n-d\,n\,\mu)\right)\right)\Big/\left(m^2(-1+\mu)^2-(-1+d)\,n\,\mu\,(1+(-2+n)\,\mu)+m\,n\,(-1+\mu)\,(1+d\,(-2+n)\,\mu)\right)$$

Out[47]= $$(m\,(n+m\,(-1+\mu)-\mu)\,(-1+\mu))\Big/\left(m^2(-1+\mu)^2-(-1+d)\,n\,\mu\,(1+(-2+n)\,\mu)+m\,n\,(-1+\mu)\,(1+d\,(-2+n)\,\mu)\right)$$

Simplify the way they are written

In[48]:=
```
QinMw - ((1 - μ) (d n μ (1 - m) + (m - μ) n - m^2 (1 - μ))) /
    (+ (d - 1) n μ (1 + (n - 2) μ) + m n (1 - μ) (1 + d (n - 2) μ) - m^2 (1 - μ)^2) // FullSimplify
```

Out[48]= 0

In[49]:=
```
QoutMw - (m (n + m (-1 + μ) - μ) (1 - μ)) /
    (+ (d - 1) n μ (1 + (n - 2) μ) + m n (1 - μ) (1 + d (n - 2) μ) - m^2 (1 - μ)^2) // FullSimplify
```

Out[49]= 0

Limit behavior

In[50]:=
```
Limit[QinMw, d → ∞] // FullSimplify
Limit[%, μ → 0] // FullSimplify
```

Out[50]= $$\frac{-1+m+\mu-m\mu}{-1+m(-2+n)(-1+\mu)-(-2+n)\mu}$$

Out[51]= $$\frac{1-m}{1+m(-2+n)}$$

In[52]:=
```
Limit[QoutMw, d → ∞]
Limit[QoutMw, μ → 0]
```

Out[52]= 0

Out[53]= 1

# Wright - Fisher

The structure of this part is the same as for the Moran version above, so comments are lighter here.

## Simplify QinM and QoutM

See Appendix B2 for details on how Qself, Qin and Qout were obtained using a formula presented in the appendix of Debarre 2017 JTB for "2D graphs".

In[54]:=

$$\text{QselfWF2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{1-(1-\mu)^2}+\frac{1}{1-(1-\mu)^2\,(\text{dself}-\text{din})^2}\,(n-1)\,d+\frac{1}{1-(1-\mu)^2\,\left(1-m-\frac{m}{d-1}\right)^2}\,(d-1)\right);$$

$$\text{QinWF2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{1-(1-\mu)^2}-\frac{1}{1-(1-\mu)^2\,(\text{dself}-\text{din})^2}\,d+\frac{1}{1-(1-\mu)^2\,\left(1-m-\frac{m}{d-1}\right)^2}\,(d-1)\right);$$

$$\text{QoutWF2} = \frac{\mu\,\lambda}{n\,d}\left(\frac{1}{1-(1-\mu)^2}-\frac{1}{1-(1-\mu)^2\,\left(1-m-\frac{m}{d-1}\right)^2}\right);$$

Find $\lambda$ using Qself == 1

In[57]:=
```
λWF = λ /. Solve[QselfWF2 == 1, λ][[1]]
```

Out[57]=

$$\frac{d\,n}{\left(\frac{1}{1-(1-\mu)^2}+\frac{d\,(-1+n)}{1-(-\text{din}+\text{dself})^2\,(1-\mu)^2}+\frac{-1+d}{1-\left(1-m-\frac{m}{-1+d}\right)^2\,(1-\mu)^2}\right)\mu}$$

Replace $\lambda$ in the equations

In[58]:=
```
QinWF = QinWF2 /. λ → λWF // FullSimplify
QoutWF = QoutWF2 /. λ → λWF // FullSimplify
```

Out[58]=

$$\frac{-\frac{d}{1-(\text{din}-\text{dself})^2\,(-1+\mu)^2}+\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{d\,(-1+n)}{1-(\text{din}-\text{dself})^2\,(-1+\mu)^2}+\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

Out[59]=

$$\frac{-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{d\,(-1+n)}{1-(\text{din}-\text{dself})^2\,(-1+\mu)^2}+\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

## Check numerically

In[60]:=
```
NGetQWF[G_, N_, graphdegree_, p_, μ_] := Module[{QT, eqs, vars, sols, QTs}, (*
  G is the dispersal graph,
  N is the size of the population,
  graphdegree is the degree of the graph (=1 in a subdivided pop),
  p is the mutation biais,
  μ is the mutation probability.
  *)

  (* Initialize the QT matrix *)
  Do[Q_{i,j} = 0; Q_{i,j} =., {i, 1, N}, {j, 1, N}];
  QT = Table[Q_{i,j}, {i, 1, N}, {j, 1, N}];
  Do[QT[[i, i]] = 1, {i, 1, N}]; (* Q_{i,i}=1 *)
  Do[QT[[i, j]] = Q_{j,i}, {j, 1, N - 1}, {i, j + 1, N}]; (* Because Q is symmetric *)

  eqs =
   Flatten[Table[Q_{i,j} == (1 - μ)^2/graphdegree^2 (Sum[G[[l, j]] G[[k, i]] QT[[k, l]], {l, 1, N}, {k, 1, N}]),
     {i, 1, N - 1}, {j, i + 1, N}]];

  vars = Flatten[Table[Q_{i,j}, {i, 1, N - 1}, {j, i + 1, N}]];
  sols = NSolve[eqs, vars];

  QTs = QT /. sols[[1]];
  QTs]
```

In[61]:=
```
prs =.;
CheckQWF[prs_, dvalues_] := Module[{NG, NQin, NQout, NQinMatrix},
  NG = ToExpression["G" <> ToString[d n /. prs] <> "generic"] /. dvalues /. prs;
  NQinMatrix = NGetQWF[NG, n d /. prs, 1, 0.5, μ /. prs];
  NQin = NQinMatrix[[1, 2]];
  NQout = NQinMatrix[[1, n d /. prs]];
  Print[{{"Qin", NQin, QinWF /. dvalues /. prs},
      {"Qout", NQout, QoutWF /. dvalues /. prs}} // Transpose // MatrixForm]
 ]
```

In[63]:=
```
CheckQWF[{m → 0.2, d → 4, n → 3, μ → 0.2}, withselfreplacement]
CheckQWF[{m → 0.2, d → 4, n → 3, μ → 0.2}, noselfreplacement]
CheckQWF[{m → 0.2, d → 2, n → 5, μ → 0.2}, withselfreplacement]
CheckQWF[{m → 0.2, d → 2, n → 5, μ → 0.2}, noselfreplacement]
```

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.218352 & 0.0816154 \\ 0.218352 & 0.0816154 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.178044 & 0.0770357 \\ 0.178044 & 0.0770357 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.17199 & 0.122413 \\ 0.17199 & 0.122413 \end{pmatrix}$$

$$\begin{pmatrix} \text{Qin} & \text{Qout} \\ 0.164772 & 0.120319 \\ 0.164772 & 0.120319 \end{pmatrix}$$

# Particular cases

## With Self Replacement

In[67]:=
```
QinWFs = QinWF /. withselfreplacement // FullSimplify
```

Out[67]=
$$\frac{-d + \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}{d(-1+n) + \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}$$

Simplify the way it is written

In[68]:=
$$\text{M1} = \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}};$$

$$\text{M2} = \frac{1}{2\mu-\mu^2};$$

$$\text{qinwfs} = \frac{-d + \text{M1} + \text{M2}}{(n-1)\,d + \text{M1} + \text{M2}};$$

```
QinWFs - qinwfs // FullSimplify
```

Out[71]= 0

In[72]:=
```
QoutWFs = QoutWF /. withselfreplacement // FullSimplify
```

Out[72]=
$$\frac{-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}{d(-1+n) + \frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}} + \frac{1}{2\mu-\mu^2}}$$

Simplify the way it is written

In[73]:=
$$\text{qoutwfs} = \frac{\frac{-1}{d-1}\text{M1} + \text{M2}}{d\,(n-1) + \text{M1} + \text{M2}}\ \text{// FullSimplify;}$$

```
qoutwfs - QoutWFs // FullSimplify
```

Out[74]= 0

Limit behavior

In[75]:=
```
Limit[QinWFs, μ → 0]
Limit[QoutWFs, μ → 0]
```

Out[75]= 1

Out[76]= 1

In[77]:=
```
Limit[QinWFs, d → ∞] // FullSimplify
Limit[%, μ → 0] // FullSimplify
```

Out[77]= $-\left(\left((-1+m)^2\,(-1+\mu)^2\right)\Big/\left(-1-2\,m\,(-1+n)\,(-1+\mu)^2+m^2\,(-1+n)\,(-1+\mu)^2+(-1+n)\,(-2+\mu)\,\mu\right)\right)$

Out[78]= $-\dfrac{(-1+m)^2}{-1+(-2+m)\,m\,(-1+n)}$

In[79]:=
```
Limit[QoutWFs, d → ∞] // FullSimplify
```

Out[79]= 0

## Comparison to Cockerham and Weir 1987

Cockerham and Weir's $\beta$

In[80]:= $dd = \left(1 - m\,\dfrac{nbdemes}{nbdemes - 1}\right)^2;$

$\rho = (1-\mu)^2;$

Update deme size, which is 2 N in their paper, to N, to adapt the result to a haploid population

In[82]:=
```
β = ρ dd / (demesize (1 - ρ dd) + ρ dd) // FullSimplify
ρ =.; dd =.;
```

Out[82]= $\left(\left(1+(-1+m)\,nbdemes\right)^2\,(-1+\mu)^2\right)\Big/\left(\left(-1+nbdemes\right)^2\left(demesize\left(1-\dfrac{\left(1+(-1+m)\,nbdemes\right)^2\,(-1+\mu)^2}{\left(-1+nbdemes\right)^2}\right)+\dfrac{\left(1+(-1+m)\,nbdemes\right)^2\,(-1+\mu)^2}{\left(-1+nbdemes\right)^2}\right)\right)$

For us:

In[84]:= $my\beta = \dfrac{QinWF - QoutWF}{1 - QoutWF}$ // FullSimplify

Out[84]= $\left(\left((-1+d)^2\left(-1+\left(din-dself\right)^2\right)+2\,(-1+d)\,d\,m-d^2\,m^2\right)(-1+\mu)^2\right)\Big/\left(-1+\left(din-dself\right)^2+2\,\mu-2\left(\left(din-dself\right)^2+n\right)\mu+\left(-1+\left(din-dself\right)^2+n\right)\mu^2+d^2\left(-1+\left(din-dself\right)^2+(-2+m)\,m\,(-1+n)+2\,\mu-2\left(\left(din-dself\right)^2+(-2+m)\,m\,(-1+n)+n\right)\mu+\left(\left(din-dself\right)^2+(-1+m)^2\,(-1+n)\right)\mu^2\right)+2\,d\left(1-\left(din-dself\right)^2-m+m\,n+2\left(-1+\left(din-dself\right)^2+m+n-m\,n\right)\mu-\left(-1+\left(din-dself\right)^2+m+n-m\,n\right)\mu^2\right)\right)$

Compare the two -> same !

In[85]:= `myβ - β /. {nbdemes → d, demesize → n} // FullSimplify`

Out[85]=

$$\begin{aligned}&(-1+\mu)^2 \Bigg(-\Bigg((1+d\,(-1+m))^2 \Big/ \\
&\quad\Bigg((-1+d)^2 \Bigg(n \Bigg(1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}\Bigg)+\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}\Bigg)\Bigg)\Bigg) + \\
&\Big((-1+d)^2 \big(-1+(\mathrm{din}-\mathrm{dself})^2\big)+2\,(-1+d)\,d\,m-d^2\,m^2\Big) \Big/ \\
&\Big(-1+(\mathrm{din}-\mathrm{dself})^2+2\,\mu-2\,\big((\mathrm{din}-\mathrm{dself})^2+n\big)\,\mu + \\
&\quad\big(-1+(\mathrm{din}-\mathrm{dself})^2+n\big)\,\mu^2+d^2\,\Big(-1+(\mathrm{din}-\mathrm{dself})^2+(-2+m)\,m\,(-1+n)+2\,\mu - \\
&\qquad 2\,\big((\mathrm{din}-\mathrm{dself})^2+(-2+m)\,m\,(-1+n)+n\big)\,\mu+\big((\mathrm{din}-\mathrm{dself})^2+(-1+m)^2\,(-1+n)\big)\,\mu^2\Big) + \\
&\quad 2\,d\,\Big(1-(\mathrm{din}-\mathrm{dself})^2-m+m\,n+2\,\big(-1+(\mathrm{din}-\mathrm{dself})^2+m+n-m\,n\big)\,\mu - \\
&\qquad \big(-1+(\mathrm{din}-\mathrm{dself})^2+m+n-m\,n\big)\,\mu^2\Big)\Big)\Bigg)\end{aligned}$$

In[86]:= `β =.; myβ =.;`

### Without Self Replacement

In[87]:= `QinWFw = QinWF /. noselfreplacement // FullSimplify`

Out[87]=

$$\frac{\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

In[88]:= `QoutWFw = QoutWF /. noselfreplacement // FullSimplify`

Out[88]=

$$\frac{-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\,\mu-\mu^2}}{\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\,\mu-\mu^2}}$$

# Export to R the Q results

Rewrite the Greek letters

In[89]:= `GreekTerms = {ω → sel, μ → mut}`

Out[89]= {ω → sel, μ → mut}

Common parts to all functions

```
In[90]:= FunctionPartB = " <- function(p, sel, mut, m, g, n, d, Idself, Ieself){
## Arguments:
#  p   mutation bias
#  sel intensity of selection
#  mut mutation probability
#  m   emigration probability
#  g   proportion of
     interactions out of the group (interaction equivalent of m)
#  n   deme size
#  d   number of demes
#  Idself whether reproduction in site where the parent is
#  Ieself whether interactions with oneself
return(
";
FunctionPartE = ")
}";
```

Function to translate Mathematica to R

```
In[92]:= ToRForm[x_] := ToString[x /. GreekTerms // CForm]
```

Do it for Q

```
In[93]:= RtxtQinM =
  "QinM " <> FunctionPartB <> ToRForm[QinM /. genericde // FullSimplify] <> FunctionPartE;
RtxtQoutM = "QoutM " <> FunctionPartB <>
   ToRForm[QoutM /. genericde // FullSimplify] <> FunctionPartE;
RtxtQinWF = "QinWF " <> FunctionPartB <>
   ToRForm[QinWF /. genericde // FullSimplify] <> FunctionPartE;
RtxtQoutWF = "QoutWF " <> FunctionPartB <>
   ToRForm[QoutWF /. genericde // FullSimplify] <> FunctionPartE;
```

Define Power function in R

```
In[97]:= PowerDef = "Power <- function(a,b) return(a^b)";
```

Combine all texts

```
In[98]:= Rtxt = PowerDef <> "

" <> RtxtQinM <> "

" <> RtxtQoutM <> "

" <> RtxtQinWF <> "

" <> RtxtQoutWF;
```

Export to txt file (Mathematica did not want R)

```
In[99]:= Export[pathtosave <> "Mathematica/analyticsQ.txt", Rtxt];
```

Convert the file extension to R

In[100]:=
```
cmd = "mv " <> pathtosave <>
    "Mathematica/analyticsQ.txt " <> pathtosave <> "Mathematica/analyticsQ.R";
```

In[101]:=
```
Get["!" <> cmd]
```

# 2) Expected Frequency Equations

## Formulas for the different life-cycles

The formulas for each term is obtained by hand, by replacing the dispersal and interaction graphs by their formulas in a subdivided population, from the equations given in Appendix B1. In some cases (e.g., $\beta$I for the Moran DB life-cycle), there is a large number of cases to consider when unpacking the sums.
D corresponds to direct / primary effects,
I corresponds to indirect / secondary effects.

### Moran, Birth-Death

#### β

In[102]:=
```
βBDD = (1 - μ) (eself + (n - 1) ein Qin + (n d - n) eout Qout);
```

In[103]:=
```
βBDI = dself eself + (n - 1) din ein + (n d - n ) dout eout (*
   *) + (n - 1) (din eself + dself ein + (n - 2) din ein + (n d - n) dout eout) Qin (*
   *) + (n d - n) (dself eout + (n - 1) din eout +
      dout eself + (n - 1) dout ein + (n d - 2 n) dout eout) Qout (*
   *) - μ/(n d) (1 + (n - 1) Qin + (n d - n) Qout) (eself + (n - 1) ein + (n d - n) eout);
```

In[104]:=
```
factorBD = p (1 - p)/μ
```

Out[104]=
$$\frac{(1-p)\,p}{\mu}$$

In[105]:=
```
βBD = factorBD (βBDD - βBDI);
```

#### γ

In[106]:=
```
γBDD = 1 - μ;
```

In[107]:=
```
γBDI = dself + (n - 1) din Qin + (n d - n) dout Qout - μ/(n d) (1 + (n - 1) Qin + (n d - n) Qout);
```

In[108]:= 
```
γBD = factorBD (γBDD - γBDI) // FullSimplify
```
Out[108]= $\frac{1}{d\,n\,\mu}(-1+p)\,p\,\left(d\,n\,\left(-1+dself+din\,(-1+n)\,Qin+(-1+d)\,dout\,n\,Qout\right)+\left(-1+Qin+n\,\left(d-Qin+Qout-d\,Qout\right)\right)\mu\right)$

## Moran, Death-Birth

### β

In[109]:= 
```
βDBD = βBDD
```
Out[109]= $\left(eself+ein\,(-1+n)\,Qin+eout\,\left(-n+d\,n\right)\,Qout\right)\,(1-\mu)$

In[110]:= 
```
βDBI = (1 - μ) (1 * dself dself eself * 1(*
        02*) + (n - 1) dself din ein * 1(*
        03*) + (n d - n) dself dout eout * 1(*
        04*) + (n - 1) dself dself ein Qin(*
        05*) + (n - 1) dself din eself Qin(*
        06*) + (n - 1) (n - 2) dself din ein Qin(*
        07*) + (n - 1) (n d - n) dself dout eout Qin(*
        08*) + (n d  - n) dself dself eout Qout(*
        09*) + (n d  - n) (n - 1) dself din eout Qout(*
        10*) + (n d  - n) dself dout eself Qout(*
        11*) + (n d  - n) (n - 1) dself dout ein Qout(*
        12*) + (n d  - n) (n d - 2 n) dself dout eout Qout(*
        13*) + (n - 1) din dself eself Qin(*
        14*) + (n - 1) din din ein Qin(*
        15*) + (n - 1) (n - 2) din din ein Qin (*
        16*) + (n - 1) (n d  - n) din dout eout Qin(*
        17*) + (n - 1) din dself ein (*
        18*) + (n - 1) din din eself(*
        19*) + (n - 1) (n - 2) din din ein(*
        20*) + (n - 1) (n d - n) din dout eout(*
        21*) + (n - 1) (n - 2) din dself ein Qin(*
        22*) + (n - 1) (n - 2) din din ein Qin(*
        23*) + (n - 1) (n - 2) din din eself Qin(*
        24*) + (n - 1) (n - 2) (n - 3) din din ein Qin(*
        25*) + (n - 1) (n - 2) (n d - n) din dout eout Qin(*
        26*) + (n - 1) (n d - n) din dself eout Qout(*
        27*) + (n - 1) (n d - n) din din eout Qout(*
        28*) + (n - 1) (n d - n) (n - 2) din din eout Qout(*
        29*) + (n - 1) (n d - n) din dout eself Qout(*
        30*) + (n - 1) (n d - n) (n - 1) din dout ein Qout(*
        31*) + (n - 1) (n d - n) (n d - 2 n) din dout eout Qout(*
        32*) + (n d - n) dout dself eself Qout(*
```

```
33*) + (n d - n) (n - 1) dout din ein Qout (*
34*) + (n d - n) dout dout eout Qout (*
35*) + (n d - n) (n - 1) dout dout eout Qout (*
36*) + (n d - n) (n d - 2 n) dout dout eout Qout (*
37*) + (n d - n) (n - 1) dout dself ein Qout (*
38*) + (n d - n) (n - 1) dout din eself Qout (*
39*) + (n d - n) (n - 1) (n - 2) dout din ein Qout (*
40*) + (n d  - n) (n - 1) dout dout eout Qout (*
41*) + (n d - n) (n - 1) (n - 1) dout dout eout Qout (*
42*) + (n d - n) (n - 1) (n d - 2 n) dout dout eout Qout (*
43*) + (n d - n) dout dself eout (*
44*) + (n d - n) (n - 1) dout din eout (*
45*) + (n d - n) dout dout eself (*
46*) + (n d - n) (n - 1) dout dout ein (*
47*) + (n d - n) (n d - 2 n) dout dout eout (*
48*) + (n d - n) (n - 1) dout dself eout Qin (*
49*) + (n d - n) (n - 1) (n - 1) dout din eout Qin (*
50*) + (n d - n) (n - 1) dout dout ein Qin (*
51*) + (n d - n) (n - 1) dout dout eself Qin (*
52*) + (n d - n) (n - 1) (n - 2) dout dout ein Qin (*
53*) + (n d - n) (n - 1) (n d - 2 n) dout dout eout Qin (*
54*) + (n d - n) (n d - 2 n) dout dself eout Qout (*
55*) + (n d - n) (n d - 2 n) (n - 1) dout din eout Qout (*
56*) + (n d - n) (n d - 2 n) dout dout eout Qout (*
57*) + (n d - n) (n d - 2 n) (n - 1) dout dout eout Qout (*
58*) + (n d - n) (n d - 2 n) dout dout eself Qout (*
59*) + (n d - n) (n d - 2 n) (n - 1) dout dout ein Qout (*
60*) + (n d - n) (n d - 2 n) (n d - 3 n) dout dout eout Qout);
```

In[111]:= **factorDB = factorBD**

Out[111]= $\frac{(1-p)\,p}{\mu}$

In[112]:= **βDB = factorDB (βDBD - βDBI) // FullSimplify**

Out[112]=
$$\frac{1}{\mu}(1-p)\,p\,\Big(\text{eself}+\text{ein}\,(-1+n)\,\text{Qin}+(-1+d)\,\text{eout}\,n\,\text{Qout}-\text{dself}^2\,\big(\text{eself}+\text{ein}\,(-1+n)\,\text{Qin}+(-1+d)\,\text{eout}\,n\,\text{Qout}\big)-\text{din}^2\,(-1+n)\,\big(\text{eself}+\text{eself}\,(-2+n)\,\text{Qin}+\text{ein}\,\big(-2+n+(3+(-3+n)\,n)\,\text{Qin}\big)+(-1+d)\,\text{eout}\,(-1+n)\,n\,\text{Qout}\big)-(-1+d)\,\text{dout}^2\,n\,\Big(\big(\text{eself}+\text{ein}\,(-1+n)+(-2+d)\,\text{eout}\,n\big)\,\big(1+(-1+n)\,\text{Qin}\big)+n\,\big((-2+d)\,\text{eself}+(-2+d)\,\text{ein}\,(-1+n)+\big(3+(-3+d)\,d\big)\,\text{eout}\,n\big)\,\text{Qout}\Big)-2\,(-1+d)\,\text{dout}\,\text{dself}\,n\,\Big(\big(\text{eself}+\text{ein}\,(-1+n)\big)\,\text{Qout}+\text{eout}\,\big(1+(-1+n)\,\text{Qin}+(-2+d)\,n\,\text{Qout}\big)\Big)-2\,\text{din}\,(-1+n)\,\Big(\text{dself}\,\big(\text{ein}+\text{eself}\,\text{Qin}+\text{ein}\,(-2+n)\,\text{Qin}+(-1+d)\,\text{eout}\,n\,\text{Qout}\big)+(-1+d)\,\text{dout}\,n\,\big(\text{eout}+\text{eout}\,(-1+n)\,\text{Qin}+\big(\text{eself}+\text{ein}\,(-1+n)+(-2+d)\,\text{eout}\,n\big)\,\text{Qout}\big)\Big)\Big)\,(1-\mu)$$

### γ

In[113]:=
```
γDBD = γBDD
```

Out[113]= $1-\mu$

In[114]:=
```
γDBI = (1 - μ) (dself^2 + (n - 1) din^2 + (n d - n) dout^2 (*
    *) + (n - 1) (dself din + din dself + (n - 2) din din + (n d - n) dout dout) Qin(*
    *) + (n d - n) (dself dout + (n - 1) din dout +
       dout dself + (n - 1) dout din + (n d - 2 n) dout dout) Qout);
```

In[115]:=
```
γDB = factorDB (γDBD - γDBI)
```

Out[115]=
$$\frac{1}{\mu}(1-p)\,p\,\Big(1-\big(\text{dself}^2+\text{din}^2\,(-1+n)+\text{dout}^2\,(-n+d\,n)+(-1+n)\,\big(2\,\text{din dself}+\text{din}^2\,(-2+n)+\text{dout}^2\,(-n+d\,n)\big)\,\text{Qin}+(-n+d\,n)\,\big(2\,\text{dout dself}+2\,\text{din dout}\,(-1+n)+\text{dout}^2\,(-2\,n+d\,n)\big)\,\text{Qout}\big)\,(1-\mu)-\mu\Big)$$

## Wright - Fisher

The formulas are the same as the Moran DB life-cycle, only the probabilities of identity by descent Q will differ.

### β

In[116]:=
```
βWFI = βDBI;
βWFD = βDBD;
βWF = βDB;
```

### γ

In[119]:=
```
γWF = γDB;
γWFD = γDBD;
γWFI = γDBI;
```

# Check the results numerically

We use generic equations valid for any life-cycle and any graph, and adapt them to our life-cycles and to a subdivided population. We compare the numerical results to the ones obtained with the equations written above.

## Full functions, any life-cycle

$\beta$ and $\gamma$ were calculated by hand - these are generic equations value for any life-cycle and any regular graph.
(see Appendix B1 for details, and Debarre 2017 JTB for even further details)

In[122]:=
```
GetBeta[sBf_, Df_, G_, GE_, Qmat_, N_, graphdegree_, Bstar_] :=
 Module[{part1, part2, factor},
  factor = p (1 - p) / (μ N Bstar);
  part1 = Sum[((1 - μ) sBf[G, N, graphdegree, j, l] - Df[G, N, graphdegree, j, l]) *
     GE[[k, l]] * Qmat[[j, k]], {j, 1, N}, {k, 1, N}, {l, 1, N}];
  factor * part1]

GetGamma[sBf_, Df_, G_, GE_, Qmat_, N_, graphdegree_, Bstar_] :=
 Module[{part1, part2, factor},
  factor = p (1 - p) / (μ N Bstar);
  part1 = Sum[((1 - μ) sBf[G, N, graphdegree, j, k] - Df[G, N, graphdegree, j, k]) *
     Qmat[[j, k]], {j, 1, N}, {k, 1, N}];
  factor *
   part1]
```

# Moran DB

## Define δB and δD

graphdegree is the degree of the graph, here equal to 1

In[124]:=
```
sBfDB[G_, N_, graphdegree_, j_, k_] :=
  (Delta[k - j] graphdegree^2 - Sum[G[[j, i]] G[[k, i]], {i, 1, N}]) / (N graphdegree^2);
DfDB[G_, N_, graphdegree_, j_, k_] := 0;
BstarDB = 1/N;
```

## β

Numerical comparison for the population of size 12

In[127]:=
```
βDBexemple = GetBeta[sBfDB, DfDB, G12generic,
   GE12generic, Q12generic, 12, 1, BstarDB /. N → 12] // FullSimplify
```

Out[127]=

$$-\frac{1}{\mu}(-1+p)\, p\, \Big((-1+dself^2)\,(eself+2\,ein\,Qin+9\,eout\,Qout)+ 2\,din^2\,(ein+eself+3\,ein\,Qin+eself\,Qin+18\,eout\,Qout)+ 18\,dout\,dself\,(eout+2\,eout\,Qin+2\,ein\,Qout+6\,eout\,Qout+eself\,Qout)+ 9\,dout^2\,\big((2\,ein+6\,eout+eself)\,(1+2\,Qin)+3\,(4\,ein+21\,eout+2\,eself)\,Qout\big)+ 4\,din\,\big(dself\,(ein+(ein+eself)\,Qin+9\,eout\,Qout)+ 9\,dout\,(eout+2\,eout\,Qin+2\,ein\,Qout+6\,eout\,Qout+eself\,Qout)\big)\Big)\,(-1+\mu)$$

In[128]:=
```
βDBexemple - βDB /. {n → 3, d → 4} // FullSimplify
```

Out[128]= 0

The difference is zero: we are fine!

Numerical comparison for the population of size 10

In[129]:=
```
βDBexemple2 = GetBeta[sBfDB, DfDB, G10generic,
   GE10generic, Q10generic, 10, 1, BstarDB /. N → 10] // FullSimplify
```

Out[129]=
$$-\frac{1}{\mu}(-1+p)\,p\,\big((-1+\mathrm{dself}^2)\,(\mathrm{eself}+4\,\mathrm{ein}\,\mathrm{Qin}+5\,\mathrm{eout}\,\mathrm{Qout})+4\,\mathrm{din}^2\,(\mathrm{eself}+3\,\mathrm{eself}\,\mathrm{Qin}+\mathrm{ein}\,(3+13\,\mathrm{Qin})+20\,\mathrm{eout}\,\mathrm{Qout})+5\,\mathrm{dout}^2\,((4\,\mathrm{ein}+\mathrm{eself})\,(1+4\,\mathrm{Qin})+25\,\mathrm{eout}\,\mathrm{Qout})+10\,\mathrm{dout}\,\mathrm{dself}\,(\mathrm{eout}+4\,\mathrm{eout}\,\mathrm{Qin}+(4\,\mathrm{ein}+\mathrm{eself})\,\mathrm{Qout})+8\,\mathrm{din}\,(\mathrm{dself}\,(\mathrm{ein}+3\,\mathrm{ein}\,\mathrm{Qin}+\mathrm{eself}\,\mathrm{Qin}+5\,\mathrm{eout}\,\mathrm{Qout})+5\,\mathrm{dout}\,(\mathrm{eout}+4\,\mathrm{eout}\,\mathrm{Qin}+4\,\mathrm{ein}\,\mathrm{Qout}+\mathrm{eself}\,\mathrm{Qout}))\big)\,(-1+\mu)$$

In[130]:=
```
βDBexemple2 - βDB /. {n → 5, d → 2} // FullSimplify
```

Out[130]= 0

The difference is zero: we are fine!

### γ

Numerical comparison for the population of size 12

In[131]:=
```
γDBexemple = GetGamma[sBfDB, DfDB, G12generic,
   GE12generic, Q12generic, 12, 1, BstarDB /. N → 12] // FullSimplify
```

Out[131]=
$$-\frac{1}{\mu}(-1+p)\,p\,\big(-1+\mathrm{dself}^2+2\,\mathrm{din}^2\,(1+\mathrm{Qin})+4\,\mathrm{din}\,(\mathrm{dself}\,\mathrm{Qin}+9\,\mathrm{dout}\,\mathrm{Qout})+9\,\mathrm{dout}\,(\mathrm{dout}+2\,\mathrm{dout}\,\mathrm{Qin}+6\,\mathrm{dout}\,\mathrm{Qout}+2\,\mathrm{dself}\,\mathrm{Qout})\big)\,(-1+\mu)$$

In[132]:=
```
γDBexemple - γDB /. {n → 3, d → 4} // FullSimplify
```

Out[132]= 0

The difference is zero: we are fine!

Numerical comparison for the population of size 10

In[133]:=
```
γDBexemple2 = GetGamma[sBfDB, DfDB, G10generic,
   GE10generic, Q10generic, 10, 1, BstarDB /. N → 10] // FullSimplify
```

Out[133]=
$$-\frac{1}{\mu}(-1+p)\,p\,\big(-1+\mathrm{dself}^2+4\,\mathrm{din}^2\,(1+3\,\mathrm{Qin})+8\,\mathrm{din}\,(\mathrm{dself}\,\mathrm{Qin}+5\,\mathrm{dout}\,\mathrm{Qout})+5\,\mathrm{dout}\,(\mathrm{dout}+4\,\mathrm{dout}\,\mathrm{Qin}+2\,\mathrm{dself}\,\mathrm{Qout})\big)\,(-1+\mu)$$

In[134]:=
```
γDBexemple2 - γDB /. {n → 5, d → 2} // FullSimplify
```

Out[134]= 0

The difference is zero: we are fine!

## Moran BD

The structure is the same as for Moran DB above - comments are lighter here!

## Define δB and δD

In[135]:=
```
sBfBD[G_, N_, graphdegree_, j_, k_] := (Delta[k - j] N - 1)/N^2;
DfBD[G_, N_, graphdegree_, j_, k_] := G[[k, j]]/(N graphdegree) - 1/N^2;
BstarBD = 1/N;
```

## Equations β

In[138]:=
```
βBDexemple = GetBeta[sBfBD, DfBD, G12generic,
   GE12generic, Q12generic, 12, 1, BstarBD /. N → 12] // FullSimplify
```

Out[138]=
$$\frac{1}{12\,\mu}(-1+p)\,p\,\Big(12\,\big((-1+\text{dself})\,(\text{eself}+2\,\text{ein Qin}+9\,\text{eout Qout})+2\,\text{din}\,(\text{ein}+(\text{ein}+\text{eself})\,\text{Qin}+9\,\text{eout Qout})+9\,\text{dout}\,(\text{eout}+2\,\text{eout Qin}+2\,\text{ein Qout}+6\,\text{eout Qout}+\text{eself Qout})\big)+(-9\,\text{eout}+11\,\text{eself}-2\,(\text{ein}-10\,\text{ein Qin}+(9\,\text{eout}+\text{eself})\,\text{Qin})-9\,(2\,\text{ein}-3\,\text{eout}+\text{eself})\,\text{Qout})\,\mu\Big)$$

Check also the different terms separately

In[139]:=
```
βIBDtest = Sum[(μ/N^2 - G[[l, j]]/N) GE[[k, l]] Q[[j, k]], {j, 1, N}, {k, 1, N}, {l, 1, N}] /.
    {G → G12generic, GE → GE12generic, Q → Q12generic, N → 12} // FullSimplify // Quiet
```

Out[139]=
$$-\text{dself}\,(\text{eself}+2\,\text{ein Qin}+9\,\text{eout Qout})-2\,\text{din}\,(\text{ein}+(\text{ein}+\text{eself})\,\text{Qin}+9\,\text{eout Qout})-9\,\text{dout}\,(\text{eout}+2\,\text{eout Qin}+2\,\text{ein Qout}+6\,\text{eout Qout}+\text{eself Qout})+\frac{1}{12}(2\,\text{ein}+9\,\text{eout}+\text{eself})\,(1+2\,\text{Qin}+9\,\text{Qout})\,\mu$$

In[140]:=
```
βIBDtest + βBDI /. {n → 3, d → 4} // FullSimplify
```

Out[140]= 0

In[141]:=
```
βDBDtest = Sum[(1 - μ)/N GE[[k, l]] Q[[l, k]], {k, 1, N}, {l, 1, N}] /.
    {G → G12generic, GE → GE12generic, Q → Q12generic, N → 12} // FullSimplify // Quiet
```

Out[141]=
$$-(\text{eself}+2\,\text{ein Qin}+9\,\text{eout Qout})\,(-1+\mu)$$

In[142]:=
```
βDBDtest - βBDD /. {n → 3, d → 4} // FullSimplify
```

Out[142]= 0

In[143]:=
```
βBDtest = (p (1 - p))/μ (βDBDtest + βIBDtest) // FullSimplify;
```

Numerical comparison (the differences are zero: we are fine!)

In[144]:=
```
βBDtest - βBD /. {n → 3, d → 4} // FullSimplify
```

Out[144]= 0

The difference is zero: we are fine!

In[145]:=
```
βBDD - βDBDtest /. {n → 3, d → 4} // FullSimplify
```

Out[145]= 0

The difference is zero: we are fine!

In[146]:=
```
βBDI + βIBDtest /. {n → 3, d → 4} // FullSimplify
```

Out[146]= 0

The difference is zero: we are fine!

## Equations γ

In[147]:=
```
γBDexemple = GetGamma[sBfBD, DfBD, G12generic,
   GE12generic, Q12generic, 12, 1, BstarBD /. N → 12] // FullSimplify
```

Out[147]= $\frac{1}{12\,\mu}(-1+p)\,p\,\left(12\,(-1+\text{dself}+2\,\text{din Qin}+9\,\text{dout Qout})+(11-2\,\text{Qin}-9\,\text{Qout})\,\mu\right)$

In[148]:=
```
γBDexemple - γBD /. {n → 3, d → 4} // Simplify
```

Out[148]= 0

The difference is zero: we are fine!

# Wright-Fisher

The structure is the same as for Moran DB above - comments are lighter here!

## Define δB and δD

In[149]:=
```
sBfWF[G_, N_, graphdegree_, j_, k_] :=
  1/graphdegree^2 (Delta[k - j] graphdegree^2 - Sum[G[[j, i]] G[[k, i]], {i, 1, N}]);
DfWF[G_, N_, graphdegree_, j_, k_] := 0;
BstarWF = 1;
```

## Equations β

In[152]:=
```
βWFexemple = GetBeta[sBfWF, DfWF, G12generic,
   GE12generic, Q12generic, 12, 1, BstarWF /. N → 12] // FullSimplify
```

Out[152]= $-\frac{1}{\mu}(-1+p)\,p\,\Big((-1+\text{dself}^2)\,(\text{eself}+2\,\text{ein Qin}+9\,\text{eout Qout})+$
$2\,\text{din}^2\,(\text{ein}+\text{eself}+3\,\text{ein Qin}+\text{eself Qin}+18\,\text{eout Qout})+$
$18\,\text{dout dself}\,(\text{eout}+2\,\text{eout Qin}+2\,\text{ein Qout}+6\,\text{eout Qout}+\text{eself Qout})+$
$9\,\text{dout}^2\,\left((2\,\text{ein}+6\,\text{eout}+\text{eself})\,(1+2\,\text{Qin})+3\,(4\,\text{ein}+21\,\text{eout}+2\,\text{eself})\,\text{Qout}\right)+$
$4\,\text{din}\,\big(\text{dself}\,(\text{ein}+(\text{ein}+\text{eself})\,\text{Qin}+9\,\text{eout Qout})+$
$9\,\text{dout}\,(\text{eout}+2\,\text{eout Qin}+2\,\text{ein Qout}+6\,\text{eout Qout}+\text{eself Qout})\big)\Big)\,(-1+\mu)$

In[153]:=
```
βWFexemple - βWF /. {n → 3, d → 4} // Simplify
```
Out[153]=
```
0
```

The difference is zero: we are fine!

## Equations γ

In[154]:=
```
γWFexemple = GetGamma[sBfWF, DfWF, G12generic,
   GE12generic, Q12generic, 12, 1, BstarWF /. N → 12] // FullSimplify
```
Out[154]=
$$-\frac{1}{\mu}(-1+p)\,p\,\left(-1+\text{dself}^2+2\,\text{din}^2\,(1+\text{Qin})+4\,\text{din}\,(\text{dself}\,\text{Qin}+9\,\text{dout}\,\text{Qout})+9\,\text{dout}\,(\text{dout}+2\,\text{dout}\,\text{Qin}+6\,\text{dout}\,\text{Qout}+2\,\text{dself}\,\text{Qout})\right)(-1+\mu)$$

In[155]:=
```
γWFexemple - γWF /. {n → 3, d → 4} // Simplify
```
Out[155]=
```
0
```

The difference is zero: we are fine!

# Expected frequencies of altruists in the population

## Moran BD

In[156]:=
```
EXBD = p + δ (βBD b - γBD c) /. {Qin → QinM, Qout → QoutM} /. genericde // Simplify
```
Out[156]=
```
(p ((b - c) (-1 + p) δ μ (m - n + Idself (-1 + m) (-1 + μ) + μ - m μ) +
    d^2 (-c Idself m n δ - c m^2 n δ + c Idself m^2 n δ + c m n^2 δ + c Idself m n p δ + c m^2 n p δ -
       c Idself m^2 n p δ - c m n^2 p δ + Idself μ - Idself m μ - n μ + 2 m n μ - m n^2 μ + 2 c Idself δ μ -
       2 c Idself m δ μ - 2 c n δ μ - c Idself n δ μ + 2 c m n δ μ + 2 c Idself m n δ μ + c m^2 n δ μ -
       c Idself m^2 n δ μ + c n^2 δ μ - 2 c m n^2 δ μ - 2 c Idself p δ μ + 2 c Idself m p δ μ +
       2 c n p δ μ + c Idself n p δ μ - 2 c m n p δ μ - 2 c Idself m n p δ μ - c m^2 n p δ μ +
       c Idself m^2 n p δ μ - c n^2 p δ μ + 2 c m n^2 p δ μ - Idself μ^2 + Idself m μ^2 + 2 n μ^2 - 2 m n μ^2 -
       n^2 μ^2 + m n^2 μ^2 - c Idself δ μ^2 + c Idself m δ μ^2 + 2 c n δ μ^2 - 2 c m n δ μ^2 - c n^2 δ μ^2 +
       c m n^2 δ μ^2 + c Idself p δ μ^2 - c Idself m p δ μ^2 - 2 c n p δ μ^2 + 2 c m n p δ μ^2 + c n^2 p δ μ^2 -
       c m n^2 p δ μ^2 + b (-1 + g) (-1 + p) δ (-Ieself (m (-1 + μ) - μ) (m + n (-1 + μ) - μ) +
          (-1 + m) n (-2 + μ) μ + Idself (-1 + m) (Ieself (m (-1 + μ) - μ) - (-2 + μ) μ))) +
    d (m^2 (-1 + μ) (-1 + c (-1 + p) δ + b (δ - p δ) + μ) +
       μ ((c + b (-1 + (-1 + g) Ieself)) (-1 + p) δ μ + n (1 + 3 c δ - 3 c p δ - 2 μ - 2 c δ μ +
             2 c p δ μ - b (-1 + p) δ (-3 + Ieself + g (2 + Ieself (-1 + μ) - μ) + μ - Ieself μ)) +
          n^2 (μ + c δ (-1 + p + μ - p μ))) + m (n (b (δ - p δ) - (-1 + c (-1 + p) δ) (-1 + μ)) -
          (-1 + p) δ μ (2 c (-1 + μ) + b (2 - Ieself - 2 μ + g (-2 + Ieself + μ)))) +
       Idself (-1 + m) (m (1 + b (-1 + p) δ + c (δ - p δ) - μ) (-1 + μ) + μ (1 - μ + c (-1 + p)
             δ (-3 + n + 2 μ) + b (-1 + p) δ (3 - Ieself - 2 μ + g (-2 + Ieself + μ)))))))) /
 (d (m^2 (-1 + μ)^2 - Idself (-1 + m) (-1 + μ) (m (-1 + μ) + μ - d μ) -
    (-1 + d) n μ (1 + (-2 + n) μ) +
    m n (-1 + μ) (1 + d (-2 + n) μ)))
```

# Moran DB

In[157]:=
```
EXDB = p + δ (βDB b - γDB c) /. {Qin → QinM, Qout → QoutM} /. genericde // Simplify
```

Out[157]=
```
(p ((-1 + n) (n - d n)^3 (-m^2 (-1 + μ)^2 + Idself (-1 + m) (-1 + μ) (m (-1 + μ) + μ - d μ) +
      (-1 + d) n μ (1 + (-2 + n) μ) - m n (-1 + μ) (1 + d (-2 + n) μ)) +
    (-1 + d)^2 n^3 (1 - p) δ (-1 + μ) (c (-1 + d) (-1 + n) (m n (2 + d (n (-1 + μ) - 2 μ)) -
        (-1 + d) (-2 + n) n μ + m^2 (-2 + d n + μ) - Idself^2 (-1 + m)^2 (d (m (-1 + μ) - μ) + μ) +
        Idself (-1 + m) (d m^2 (-1 + μ) + 2 (-1 + d) μ + m (2 - 2 d μ))) +
      b (2 m^2 - 2 d m^2 - 2 d Ieself m^2 + 2 d^2 Ieself m^2 + 2 d g Ieself m^2 - 2 d^2 g Ieself m^2 +
        d g m^3 + d Ieself m^3 - d^2 Ieself m^3 - d g Ieself m^3 + d^2 g Ieself m^3 - 2 m n + 2 d m n +
        2 d Ieself m n - 2 d^2 Ieself m n - 2 d g Ieself m n + 2 d^2 g Ieself m n - 2 m^2 n +
        3 d m^2 n - d^2 m^2 n - 2 d g m^2 n + d^2 g m^2 n - d m^3 n + d^2 m^3 n - d^2 g m^3 n + 2 m n^2 - 3 d m n^2 +
        d^2 m n^2 + d g m n^2 - d^2 g m n^2 - d Ieself m n^2 + d^2 Ieself m n^2 + d g Ieself m n^2 -
        d^2 g Ieself m n^2 + d m^2 n^2 - d^2 m^2 n^2 + d^2 g m^2 n^2 + 2 g m μ - 2 d g m μ + 2 Ieself m μ -
        4 d Ieself m μ + 2 d^2 Ieself m μ - 2 g Ieself m μ + 4 d g Ieself m μ - 2 d^2 g Ieself m μ -
        m^2 μ + d m^2 μ - g m^2 μ + 2 d g m^2 μ - Ieself m^2 μ + 4 d Ieself m^2 μ - 3 d^2 Ieself m^2 μ +
        g Ieself m^2 μ - 4 d g Ieself m^2 μ + 3 d^2 g Ieself m^2 μ - d g m^3 μ - d Ieself m^3 μ +
        d^2 Ieself m^3 μ + d g Ieself m^3 μ - d^2 g Ieself m^3 μ + 2 n μ - 4 d n μ + 2 d^2 n μ -
        2 g n μ + 4 d g n μ - 2 d^2 g n μ - 2 Ieself n μ + 4 d Ieself n μ - 2 d^2 Ieself n μ +
        2 g Ieself n μ - 4 d g Ieself n μ + 2 d^2 g Ieself n μ - 2 m n μ + 6 d m n μ - 4 d^2 m n μ -
        4 d g m n μ + 4 d^2 g m n μ - 2 d Ieself m n μ + 2 d^2 Ieself m n μ + 2 d g Ieself m n μ -
        2 d^2 g Ieself m n μ + 2 m^2 n μ - 5 d m^2 n μ + 3 d^2 m^2 n μ + 2 d g m^2 n μ - 3 d^2 g m^2 n μ +
        d m^3 n μ - d^2 m^3 n μ + d^2 g m^3 n μ - n^2 μ + 2 d n^2 μ - d^2 n^2 μ + g n^2 μ - 2 d g n^2 μ + d^2 g n^2 μ +
        Ieself n^2 μ - 2 d Ieself n^2 μ + d^2 Ieself n^2 μ - g Ieself n^2 μ + 2 d g Ieself n^2 μ -
        d^2 g Ieself n^2 μ - d m n^2 μ + d^2 m n^2 μ + d g m n^2 μ - d^2 g m n^2 μ + d Ieself m n^2 μ -
        d^2 Ieself m n^2 μ - d g Ieself m n^2 μ + d^2 g Ieself m n^2 μ - (-1 + d) (-1 + g)
         Idself^2 (-1 + Ieself) (-1 + m)^2 (d (m (-1 + μ) - μ) + μ) + Idself (-1 + m)
         (d m^2 (-1 - 2 g (-1 + Ieself) + 2 Ieself + d (-1 + g) (-1 + 2 Ieself - n) - n) (-1 +
              μ) + 2 (-1 + d)^2 (-1 + g) (-1 + Ieself) μ - 2 (-1 + d) m (-1 + n + g μ + Ieself μ -
              g Ieself μ - n μ - d (-1 + g) (Ieself + n (-1 + μ) + μ - 2 Ieself μ))))))) /
 ((-1 + n) (n - d n)^3 (-m^2 (-1 + μ)^2 + Idself (-1 + m) (-1 + μ) (m (-1 + μ) + μ - d μ) +
     (-1 + d)
      n
      μ
      (1 + (-2 + n) μ) - m
      n
      (-1 + μ)
      (1 + d (-2 + n) μ)))
```

# Wright - Fisher

In[158]:=
```
EXWF = p + δ (βWF b - γWF c) /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify
```

Out[158]=

$$p+\frac{1}{\mu}(1-p)\,p\,\delta\,\Bigg(-c\,\Bigg(1-\mu-(1-\mu)\,\Bigg(\frac{\text{Idself}^2\,(-1+m)^2}{n^2}+\frac{(-1+m)^2\,(\text{Idself}-n)^2}{(-1+n)\,n^2}+$$

$$\frac{m^2}{(-1+d)\,n}-\Bigg(\big(2+d\,(-2+m)\big)\,m\,\Bigg(-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/$$

$$\Bigg((-1+d)\,\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)+$$

$$\Bigg(\Big(2\,(-1+d)\,\text{Idself}\,(-1+m)^2-(-1+d)\,\text{Idself}^2\,(-1+m)^2+(-1+d)$$

$$(-2+n)\,n-2\,(-1+d)\,m\,(-2+n)\,n+m^2\,\big(1+d\,(-2+n)\,n\big)\Big)$$

$$\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/\Bigg((-1+d)$$

$$(-1+n)\,n\,\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg)\Bigg)\Bigg)+$$

$$b\,(1-\mu)\,\Bigg(\frac{\text{Ieself}-g\,\text{Ieself}}{n}+\Bigg(g\,\Bigg(-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/$$

$$\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)-$$

$$\Bigg((1-g)\,(\text{Ieself}-n)\,\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/$$

$$\Bigg(n\,\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)-\frac{1}{n^2}$$

$$(\text{Idself}-\text{Idself}\,m)^2\,\Bigg(\frac{\text{Ieself}-g\,\text{Ieself}}{n}+\Bigg(g\,\Bigg(-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/$$

$$\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)-\Bigg((1-g)$$

$$(\text{Ieself}-n)\,\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/$$

$$\Bigg(n\,\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg)-$$

$$\frac{1}{(-1+d)^2\,n}m^2\,\Bigg(\Bigg(\big(2-3\,d+d^2+g\big)\,n\,\Bigg(-\frac{1}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/$$

$$\Bigg(\frac{-1+d}{1-\frac{(1+d\,(-1+m))^2\,(-1+\mu)^2}{(-1+d)^2}}+\frac{d\,(-1+n)}{1-\frac{(-1+\text{Idself})^2\,(-1+m)^2\,(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)+(-1+d-g)$$

$$(1+((-1+n)(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))) -$$

$$\frac{1}{n}2\,\text{Idself}\,(1-m)\,m\,(((1-g)(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}) +$$

$$\frac{1}{(-1+d)\,n}g\,(1+((-2+d)\,n\,(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}) +$$

$$((-1+n)(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))) -$$

$$\frac{1}{(-1+n)\,n^2}(-1+m)^2(\text{Idself}-n)^2(\frac{\text{Ieself}-g\,\text{Ieself}}{n}+$$

$$(g(-1+n)(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}) + ((1-g)\,\text{Ieself}$$

$$(-2+n)(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(n(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2})) +$$

$$\frac{1}{n-n^2}(1-g)(\text{Ieself}-n)(-2+n+((3+(-3+n)\,n)$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\text{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}))) + \frac{1}{n}$$

$$2(1-m)(\text{Idself}-n)(\frac{1}{-1+d}m(\frac{g}{n}+((-1+d-g)(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}))/$$

$$\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)+$$

$$\Bigg(g(-1+n)\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\Bigg)\Bigg/$$

$$\Bigg(n\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\Bigg)\Bigg)\Bigg)+\frac{1}{n}$$

$$\mathrm{Idself}\,(1-m)\Bigg(\frac{(-1+g)(\mathrm{Ieself}-n)}{(-1+n)\,n}+\Bigg(g\left(-\frac{1}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{1}{2\mu-\mu^2}\right)\Bigg)\Bigg/$$

$$\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)+\Bigg((1-g)$$

$$\mathrm{Ieself}\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-\frac{d}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\Bigg)\Bigg/$$

$$\Bigg(n\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\Bigg)+$$

$$\Bigg((1-g)(\mathrm{Ieself}-n)(-2+n)\Bigg(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}-$$

$$\frac{d}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\Bigg)\Bigg)\Bigg/\Bigg((n-n^2)$$

$$\left(\frac{-1+d}{1-\frac{(1+d(-1+m))^2(-1+\mu)^2}{(-1+d)^2}}+\frac{d(-1+n)}{1-\frac{(-1+\mathrm{Idself})^2(-1+m)^2(-1+\mu)^2}{(-1+n)^2}}+\frac{1}{2\mu-\mu^2}\right)\Bigg)\Bigg)\Bigg)\Bigg)\Bigg)\Bigg)$$

# Export to R

## Export the EX formulas to R

Rewrite the Greek letters

In[159]:=

```
GreekTerms = {δ → sel, μ → mut}
```

Out[159]=

```
{δ → sel, μ → mut}
```

Common parts to all functions

```
In[160]:= FunctionPartB = " <- function(b, c, p, sel, mut, m, g, n, d, Idself, Ieself){
## Arguments:
#  b   benefit of interaction
#  c   cost of interaction
#  p   mutation bias
#  sel intensity of selection
#  mut mutation probability
#  m   emigration probability
#  g   proportion of
     interactions out of the group (interaction equivalent of m)
#  n   deme size
#  d   number of demes
#  Idself whether reproduction in site where the parent is
#  Ieself whether interactions with oneself
return(
";
FunctionPartE = ")
}";
```

Function to translate Mathematica to R

```
In[162]:= ToRForm[x_] := ToString[x /. GreekTerms // CForm]
```

Do it for all life cycles

```
In[163]:= RtxtBD = "pBD " <> FunctionPartB <> ToRForm[EXBD] <> FunctionPartE;
RtxtDB = "pDB " <> FunctionPartB <> ToRForm[EXDB] <> FunctionPartE;
RtxtWF = "pWF " <> FunctionPartB <> ToRForm[EXWF] <> FunctionPartE;
```

Define Power function in R

```
In[166]:= PowerDef = "Power <- function(a,b) return(a^b)";
```

Combine all texts

```
In[167]:= Rtxt = PowerDef <> "

" <> RtxtBD <> "

" <> RtxtDB <> "

" <> RtxtWF;
```

Export to txt file (Mathematica did not want R)

```
In[168]:= Export[pathtosave <> "Mathematica/analytics.txt", Rtxt];
```

Convert the file extension to R

```
In[169]:= cmd = "mv" <> " " <> pathtosave <>
    "Mathematica/analytics.txt " <> pathtosave <> "Mathematica/analytics.R";
```

```
In[170]:= Get["!" <> cmd];
```

## Export to R the $\beta$ and $\gamma$ functions

Rewrite the Greek letters

In[171]:=
```
GreekTerms = {ω → sel, μ → mut}
```

Out[171]=
```
{ω → sel, μ → mut}
```

Common parts to all functions

In[172]:=
```
FunctionPartB = " <- function(p, sel, mut, m, g, n, d, Idself, Ieself){
## Arguments:
#  p   mutation bias
#  sel intensity of selection
#  mut mutation probability
#  m   emigration probability
#  g   proportion of
     interactions out of the group (interaction equivalent of m)
#  n   deme size
#  d   number of demes
#  Idself whether reproduction in site where the parent is
#  Ieself whether interactions with oneself
return(
";
FunctionPartE = ")
}";
```

Function to translate Mathematica to R

In[174]:=
```
ToRForm[x_] := ToString[x /. GreekTerms // CForm]
```

Do it for $\beta$ and $\gamma$

In[175]:=
```
RtxtbBDD = "bBDD " <> FunctionPartB <> ToRForm[
    βBDD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
RtxtbBDI = "bBDI " <> FunctionPartB <> ToRForm[
    βBDI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
RtxtcBDD = "cBDD " <> FunctionPartB <> ToRForm[
    γBDD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
RtxtcBDI = "cBDI " <> FunctionPartB <> ToRForm[
    γBDI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
```

In[179]:=

```
RtxtbDBD = "bDBD " <> FunctionPartB <> ToRForm[
    βDBD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
RtxtbDBI = "bDBI " <> FunctionPartB <> ToRForm[
    βDBI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
RtxtcDBD = "cDBD " <> FunctionPartB <> ToRForm[
    γDBD /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
RtxtcDBI = "cDBI " <> FunctionPartB <> ToRForm[
    γDBI /. {Qin → QinM, Qout → QoutM} /. genericde // FullSimplify] <> FunctionPartE;
```

In[183]:=

```
RtxtbWFD = "bWFD " <> FunctionPartB <> ToRForm[
    βWFD /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
RtxtbWFI = "bWFI " <> FunctionPartB <> ToRForm[
    βWFI /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
RtxtcWFD = "cWFD " <> FunctionPartB <> ToRForm[
    γWFD /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
```

In[186]:=

```
RtxtcWFI = "cWFI " <> FunctionPartB <> ToRForm[
    γWFI /. {Qin → QinWF, Qout → QoutWF} /. genericde // Simplify] <> FunctionPartE;
```

Define Power function in R

In[187]:=

```
PowerDef = "Power <- function(a,b) return(a^b)";
```

Combine all texts

In[188]:=
```
Rtxt = PowerDef <> "

" <> RtxtbBDD <> "

" <> RtxtbBDI <> "

" <> RtxtcBDD <> "

" <> RtxtcBDI <> "

" <> RtxtbDBD <> "

" <> RtxtbDBI <> "

" <> RtxtcDBD <> "

" <> RtxtcDBI <> "

" <> RtxtbWFD <> "

" <> RtxtbWFI <> "

" <> RtxtcWFD <> "

" <> RtxtcWFI;
```

Export to txt file (Mathematica did not want R)

In[189]:=
```
Export[pathtosave <> "Mathematica/analyticsBC.txt", Rtxt];
```

Convert the file extension to R

In[190]:=
```
cmd = "mv " <> pathtosave <> "Mathematica/analyticsBC.txt " <>
  pathtosave <> "Mathematica/analyticsBC.R"
```

Out[190]=
```
mv
  ~/Documents/Work/Projects/2016_SocEvolSubdivPop/Programs/Mathematica/analyticsBC.
  txt
  ~/Documents/Work/Projects/2016_SocEvolSubdivPop/Programs/Mathematica/analyticsBC.R
```

In[191]:=
```
Get["!" <> cmd];
```

# 3) Changes with *m*

Here, we consider a population structure
- with no social interactions with oneself (eself = 0),

- such that an offspring can establish at the very site of its parent and replace it (dself > 0),
- with social interactions strictly within a deme (g = 0).
(i.e., the structure used in the manuscript).

In[192]:=
```
mychange = genericde /. Idself → 1 /. Ieself → 0 /. g → 0;
ApplyParms[x_] := x /. genericde /. Idself → 1 /. Ieself → 0 /. g → 0
```

Parameters values used in figure 2

In[194]:=
```
figparms = {b → 15, c → 1, d → 15, n → 4, p → 0.45, δ → 0.005};
```

# Probabilities of identity by descent change with the emigration probability *m*

## Moran: derivatives with respect to *m*

### Qin

In[195]:=
```
D[QinMs, m] // FullSimplify
```

Out[195]=
$$\frac{(-1+d)^2\, n\, (-1+\mu)\, \mu^2}{\left((-1+d)\, \mu\, (1+(-1+n)\, \mu) - m\, (-1+\mu)\, (1+d\, (-1+n)\, \mu)\right)^2}$$

-> The slope is negative for Qin

### Qout

In[196]:=
```
D[QoutMs, m] // FullSimplify
```

Out[196]=
$$-\frac{(-1+d)\, (-1+\mu)\, \mu\, (1+(-1+n)\, \mu)}{\left((-1+d)\, \mu\, (1+(-1+n)\, \mu) - m\, (-1+\mu)\, (1+d\, (-1+n)\, \mu)\right)^2}$$

-> It is positive for Qout.

## Wright - Fisher: derivatives with respect to *m*

### Qin

In[197]:=
```
dqw = D[QinWFs, m] // FullSimplify
```

Out[197]=
$$\left(2\, (-1+d)^3\, (1+d\, (-1+m))\, n\, (-2+\mu)^2\, (-1+\mu)^2\, \mu^2\right) \Big/ \left((-1+d)^2\, (-2+\mu)\, \mu\, (-1+(-1+n)\, (-2+\mu)\, \mu) - 2\, (-1+d)\, m\, (-1+\mu)^2\, (-1+d\, (-1+n)\, (-2+\mu)\, \mu) + d\, m^2\, (-1+\mu)^2\, (-1+d\, (-1+n)\, (-2+\mu)\, \mu)\right)^2$$

In[198]:= `Solve[dqw == 0, m] // FullSimplify`

Out[198]= $\left\{\left\{m \to \frac{-1+d}{d}\right\}\right\}$

In[199]:= `dqw /. m → 0 // FullSimplify`
`dqw /. m → 1 // FullSimplify`

Out[199]= $-\frac{2\, n\, (-1+\mu)^2}{(-1+(-1+n)\, (-2+\mu)\, \mu)^2}$

Out[200]= $\left(2\, (-1+d)^3\, n\, \mu^2\, \left(2-3\,\mu+\mu^2\right)^2\right) \Big/ \left(-2-d^2\, n\, (-2+\mu)\, \mu+(-2+\mu)\, \mu\, (-3+(-1+n)\, (-2+\mu)\, \mu)+d\, (1+(1+2\, n)\, (-2+\mu)\, \mu)\right)^2$

–> The slope of QinWF is negative until mc = $\frac{d-1}{d}$, positive after.

### Qout

In[201]:= `dqwo = D[QoutWFs, m] // FullSimplify`

Out[201]= $-\left(\left(2\, (-1+d)^2\, (1+d\, (-1+m))\, (-2+\mu)\, (-1+\mu)^2\, \mu\, (-1+(-1+n)\, (-2+\mu)\, \mu)\right) \Big/ \left(\left(-1+d\right)^2\, (-2+\mu)\, \mu\, (-1+(-1+n)\, (-2+\mu)\, \mu)-2\, (-1+d)\, m\, (-1+\mu)^2\, (-1+d\, (-1+n)\, (-2+\mu)\, \mu)+d\, m^2\, (-1+\mu)^2\, (-1+d\, (-1+n)\, (-2+\mu)\, \mu)\right)^2\right)$

In[202]:= `Solve[dqwo == 0, m] // FullSimplify`

Out[202]= $\left\{\left\{m \to \frac{-1+d}{d}\right\}\right\}$

-> And it is the opposite for QoutWF.

# BD life-cycle, changes with *m*

## Indirect/secondary term

In[203]:= `bi = βBDI /. {Qin → QinM, Qout → QoutM} /. mychange // FullSimplify`

Out[203]= $\frac{-d\, m+d\, (1+d\, (-1+m)+m)\, \mu+(-1+d-d\, m)\, \mu^2}{d\, \left(-(-1+d)\, \mu\, (1+(-1+n)\, \mu)+m\, (-1+\mu)\, (1+d\, (-1+n)\, \mu)\right)}$

Derivative with respect to *m*

In[204]:= `dbi = D[bi, m] // FullSimplify`

Out[204]= $-\left(\left((-1+d)^2\, (1+d\, n-\mu)\, \mu^2\right) \Big/ \left(d\, \left((-1+d)\, \mu\, (1+(-1+n)\, \mu)-m\, (-1+\mu)\, (1+d\, (-1+n)\, \mu)\right)^2\right)\right)$

-> $\beta$BDI decreases with m

## EX

In[205]:= `ex = EXBD // ApplyParms // FullSimplify`

Out[205]=
$$-\Big(\Big(p\,\big((b-c)\,(-1+p)\,\delta\,\mu+d\,\big(m+(b-c)\,m\,(-1+p)\,\delta-\big(1+m+(3\,b+c\,(-3+n))\,(-1+p)\,\delta\big)\,\mu+\big(1-n+(b+c\,(-1+n))\,(-1+p)\,\delta\big)\,\mu^2\big)+d^2\,\big(c\,m\,n\,(-1+p)\,\delta+\mu+\big(m\,(-1+n)-\big(2\,(b-c)\,(-1+m)+c\,(-1+2\,m)\,n\big)\,(-1+p)\,\delta\big)\,\mu+(-1+m)\,\big(1-n+(b+c\,(-1+n))\,(-1+p)\,\delta\big)\,\mu^2\big)\big)\Big)\Big/\Big(d\,\big(-(-1+d)\,\mu\,(1+(-1+n)\,\mu)+m\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)\big)\Big)\Big)$$

Derivative with respect to *m*

In[206]:= `dex = D[ex, m] // FullSimplify`

Out[206]=
$$\Big((-1+d)^2\,(-1+p)\,p\,\delta\,\mu\,\big(c+c\,d\,n-c\,\mu+b\,\big(-1+(1+d\,n\,(-2+\mu))\,\mu\big)\big)\Big)\Big/\Big(d\,\big((-1+d)\,\mu\,(1+(-1+n)\,\mu)-m\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)\big)^2\Big)$$

In[207]:= `Solve[dex == 0, m]`

Out[207]= { }

EX is a monotonic function of m. Is it increasing or decreasing? This depends on $\mu$:

In[208]:= `muc = Solve[dex == 0, μ] // FullSimplify`

Out[208]=
$$\Big\{\{\mu\to 0\},\ \Big\{\mu\to -\frac{b-c-2\,b\,d\,n+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n}\Big\},\ \Big\{\mu\to\frac{c+b\,(-1+2\,d\,n)+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n}\Big\}\Big\}$$

We want the middle one (admissibility)

In[209]:= `mucBD = μ /. muc[[2]] // FullSimplify;`

Simplify in more human form

In[210]:=
$$1-\frac{b-c+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n}-\text{mucBD // FullSimplify}$$

Out[210]= 0

In[211]:= `dex /. μ → 1 // FullSimplify`

Out[211]=
$$-\frac{(b-c)\,(-1+p)\,p\,\delta}{n}$$

-> E[$\overline{X}$] is increasing with *m* when $\mu > \mu_c^{BD}$, decreasing otherwise (but flat when $\mu = 0$).

$$\mu_c^{BD}=1-\frac{b-c+\sqrt{(b-c)\,(b-c+4\,b\,d^2\,n^2)}}{2\,b\,d\,n}$$

# DB life - cycle, changes with *m*

## Indirect/secondary term

In[248]:= `bi = βDBI /. {Qin → QinM, Qout → QoutM} /. mychange // FullSimplify`

Out[248]=
$$\frac{(-1+\mu)\left(m+\left(1+d\,(-1+m)\right)(-1+m)\,\mu\right)}{-\left(-1+d\right)\mu\,\left(1+(-1+n)\,\mu\right)+m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)}$$

Derivative with respect to *m*

In[249]:= `dbi = D[bi, m] // FullSimplify`

Out[249]=
$$\Big((-1+\mu)\Big(-(-1+\mu)\left(m+\left(1+d\,(-1+m)\right)(-1+m)\,\mu\right)\left(1+d\,(-1+n)\,\mu\right)+\left(1+\left(1+2\,d\,(-1+m)\right)\mu\right)\left(-\left(-1+d\right)\mu\,\left(1+(-1+n)\,\mu\right)+m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)\right)\Big)\Big)\Big/\left(\left(-1+d\right)\mu\,\left(1+(-1+n)\,\mu\right)-m\,(-1+\mu)\left(1+d\,(-1+n)\,\mu\right)\right)^2$$

In[250]:= `mcs = Solve[dbi == 0, m] // FullSimplify`

Out[250]=
$$\Big\{\Big\{m\to\Big(\left(-1+d\right)d\,(-1+\mu)\,\mu\,\left(1+(-1+n)\,\mu\right)-\sqrt{\left(\left(-1+d\right)^2 d\,(-1+\mu)^2\,\mu\,\left(n+(-1+\mu)^2+d\,n^2\,\mu-n\,\mu^2\right)\right)}\Big)\Big/\left(d\,(-1+\mu)^2\left(1+d\,(-1+n)\,\mu\right)\right)\Big\},$$
$$\Big\{m\to\Big(\left(-1+d\right)d\,(-1+\mu)\,\mu\,\left(1+(-1+n)\,\mu\right)+\sqrt{\left(\left(-1+d\right)^2 d\,(-1+\mu)^2\,\mu\,\left(n+(-1+\mu)^2+d\,n^2\,\mu-n\,\mu^2\right)\right)}\Big)\Big/\left(d\,(-1+\mu)^2\left(1+d\,(-1+n)\,\mu\right)\right)\Big\}\Big\}$$

the first term is negative -> not solution

In[251]:= `mcDBI = m /. mcs[[2]];`

Second derivative of *β*DBI with respect to *m*, evaluated at the critical value of *m*

In[252]:= `D[bi, m, m] /. m → mcDBI // FullSimplify`

Out[252]=
$$\frac{2\,d^2\,(-1+\mu)^2\,\mu}{\sqrt{\left(-1+d\right)^2 d\,(-1+\mu)^2\,\mu\,\left(n+(-1+\mu)^2+d\,n^2\,\mu-n\,\mu^2\right)}}$$

-> It is a minimum.

In[253]:= `Limit[mcDBI, μ → 0]`

Out[253]= 0

Compare the value of this critical value to $\frac{d-1}{d}$

In[254]:= `ddd = mcDBI - (d - 1)/d // FullSimplify`

Out[254]=
$$\frac{1}{d}\left(1+\left(\sqrt{(-1+d)^2\,d\,(-1+\mu)^2\,\mu\,\left(n+(-1+\mu)^2+d\,n^2\,\mu-n\,\mu^2\right)}+d\,(-1+\mu)\,\left(1+\mu\,\left(-2+d\,n+\mu-n\,\mu\right)\right)\right)\Big/\left((-1+\mu)^2\,\left(1+d\,(-1+n)\,\mu\right)\right)\right)$$

In[255]:= `Solve[ddd == 0]`

Out[255]= $\{\{d \to 1\}\}$

ddd has a constant sign (because d > 1); it is the same sign as when $\mu$->0, i.e. it is negative

This means that mcDBI < $\frac{d-1}{d}$.

–> $\beta$DBI reaches a minimum when m = mcDBI,

with mcDBI < $\frac{d-1}{d}$. In the manuscript, we denote mcDBI by the symbol $m_c$'.

The formula for $m_c$' is

In[256]:= `mcDBI`

Out[256]=
$$\left((-1+d)\,d\,(-1+\mu)\,\mu\,(1+(-1+n)\,\mu)+\sqrt{(-1+d)^2\,d\,(-1+\mu)^2\,\mu\,\left(n+(-1+\mu)^2+d\,n^2\,\mu-n\,\mu^2\right)}\right)\Big/\left(d\,(-1+\mu)^2\,\left(1+d\,(-1+n)\,\mu\right)\right)$$

## EX

In[257]:= `ex = EXDB // ApplyParms // FullSimplify`

Out[257]=
$$\Big(p\,\Big((b-c)\,d\,m^2\,(-1+p)\,\delta\,(-1+\mu)+(-1+d)\,\mu\,\left(-1+b\,(-1+p)\,\delta\,(-1+\mu)+c\,(-1+n)\,(-1+p)\,\delta\,(-1+\mu)+\mu-n\,\mu\right)-m\,(-1+\mu)\,\left(-1+d\,\mu-d\,n\,\mu+b\,(-1+p)\,\delta\,\left(-2+d+d\,\mu\right)+c\,(-1+p)\,\delta\,\left(2-d\,(1+n)+d\,(-1+n)\,\mu\right)\right)\Big)\Big)\Big/\left(-(-1+d)\,\mu\,(1+(-1+n)\,\mu)+m\,(-1+\mu)\,\left(1+d\,(-1+n)\,\mu\right)\right)$$

Derivative with respect to *m*

In[258]:=
```
dex = D[ex, m] // FullSimplify
```

Out[258]=
$$\Big((-1+p)\,p\,\delta\,(-1+\mu)\Big(c\,\Big(d\,m^2+\big(-1-n+d\,(-m\,(2+m)+2\,(1+n)+d\,(-1+m)\,(1+m\,(-1+n)+n))\big)\,\mu-(1+d\,(-1+m))^2\,(-1+n)\,\mu^2\Big)+b\,\Big(-d\,m^2+\mu+d\,\big(-2+m\,(2+m)+d\,(1+m\,(-2+m-m\,n))\big)\,\mu+\Big(-(1+d\,(-1+m))^2+\big(2+2\,d\,(-2+m)+d^2\,(2+(-2+m)\,m)\big)\,n\Big)\,\mu^2\Big)\Big)\Big)\Big/\Big(\big((-1+d)\,\mu\,(1+(-1+n)\,\mu)-m\,(-1+\mu)\,(1+d\,(-1+n)\,\mu)\big)^2\Big)$$

Here the formulas are more complicated, so we first concentrate on the initial changes, i.e., on changes when *m* is close to 0.

## Initial increase?

Close to m=0

In[259]:=
```
dex0 = dex /. m → 0 // FullSimplify
```

Out[259]=
$$\big((-1+p)\,p\,\delta\,(-1+\mu)\,(b+b\,(-1+2\,n)\,\mu-c\,(1+n+(-1+n)\,\mu))\big)\Big/\big(\mu\,(1+(-1+n)\,\mu)^2\big)$$

In[260]:=
```
muc0 = Solve[dex0 == 0, μ] // FullSimplify
```

Out[260]=
$$\left\{\{\mu\to 1\},\left\{\mu\to\frac{-b+c+c\,n}{c-c\,n+b\,(-1+2\,n)}\right\}\right\}$$

In[261]:=
```
Solve[(μ /. muc0[[2]]) == 0, b]
```

Out[261]=
$$\{\{b\to c\,(1+n)\}\}$$

In[262]:=
```
Limit[(μ /. muc0[[2]]), b → ∞]
```

Out[262]=
$$\frac{1}{1-2\,n}$$

In[263]:=
```
Limit[dex0, μ → 0]
```

Out[263]=
```
(-1 + p) p δ DirectedInfinity[-b + c + c n]
```

-> E[$\overline{X}$] increases initially with *m* if $\mu$>$\mu$c, and $\mu$c<0 if b>c(n+1).

In other words,
- if b>c(n+1), E[$\overline{X}$] increases initially with *m* for any value of $\mu$,
- otherwise, E[$\overline{X}$] increases initially with *m* if $\mu > \mu_c^{BD}$, with $\mu_c^{BD} = \frac{-b+c+c\,n}{c-c\,n+b\,(-1+2\,n)}$.

With our parameters, b>c(n+1).

## Maximum value of EX

In[264]:=
```
mcs = Solve[dex == 0, m] // FullSimplify
```

Out[264]=

$$\left\{\left\{m \to \frac{1}{(-1+\mu)^2\,(1+d\,(-1+n)\,\mu)}\left((-1+d)\,(-1+\mu)\,\mu\,(1+(-1+n)\,\mu) - \frac{1}{(b-c)\,d\,(-1+p)}\left(\sqrt{\left(-(b-c)\,(-1+d)^2\,d\,(-1+p)^2\,(-1+\mu)^2\,\mu\,\left(c\,(1+n)+c\,\mu\,\left(-2+d\,n^2+\mu-n\,\mu\right)+b\,\left(-1+\mu\,\left(2-\mu+n\,\left(2\,(-1+\mu)+d\,(-1+(-1+n)\,(-2+\mu)\,\mu)\right)\right)\right)\right)\right)}\right)\right)\right\},\right.$$

$$\left.\left\{m \to \frac{1}{(-1+\mu)^2\,(1+d\,(-1+n)\,\mu)}\left((-1+d)\,(-1+\mu)\,\mu\,(1+(-1+n)\,\mu) + \frac{1}{(b-c)\,d\,(-1+p)}\left(\sqrt{\left(-(b-c)\,(-1+d)^2\,d\,(-1+p)^2\,(-1+\mu)^2\,\mu\,\left(c\,(1+n)+c\,\mu\,\left(-2+d\,n^2+\mu-n\,\mu\right)+b\,\left(-1+\mu\,\left(2-\mu+n\,\left(2\,(-1+\mu)+d\,(-1+(-1+n)\,(-2+\mu)\,\mu)\right)\right)\right)\right)\right)}\right)\right)\right\}\right\}$$

-> The admissible solution is the first one (note (-1+p) at the denominator).
This is the critical value of *m*, at which the maximum value of E[$\overline{X}$] is attained. Its formula is given by

In[265]:=
```
mmax = m /. mcs[[1]] // FullSimplify
```

Out[265]=

$$\frac{1}{(-1+\mu)^2\,(1+d\,(-1+n)\,\mu)}\left((-1+d)\,(-1+\mu)\,\mu\,(1+(-1+n)\,\mu) - \frac{1}{(b-c)\,d\,(-1+p)}\left(\sqrt{\left(-(b-c)\,(-1+d)^2\,d\,(-1+p)^2\,(-1+\mu)^2\,\mu\,\left(c\,(1+n)+c\,\mu\,\left(-2+d\,n^2+\mu-n\,\mu\right)+b\,\left(-1+\mu\,\left(2-\mu+n\,\left(2\,(-1+\mu)+d\,(-1+(-1+n)\,(-2+\mu)\,\mu)\right)\right)\right)\right)\right)}\right)\right)$$

Numerically check threshold value, i.e. that = 0 when $\mu$ = muc (formula too big to simplify...)

In[325]:=
```
myRound[x_] := Round[x, 12];
mmax /. muc0[[2]] /. {b → 2, c → 1, d → 15, n → 4, p → 0.45} // myRound
mmax /. muc0[[2]] /. {b → 4, c → 1, d → 15, n → 4, p → 0.45} // myRound
mmax /. muc0[[2]] /. {b → 4, c → 1, d → 2, n → 4, p → 0.45}
```

Out[326]= 0

Out[327]= 0

Out[328]= 0.

Check value for low mutation

In[266]:=
```
Limit[mmax, μ → 0]
```

Out[266]= 0

-> When $\mu$ -> 0, this argmax gets closer to 0: at the limit, E[$\overline{X}$] just decreases with *m.*

With our parameters,

In[267]:=
```
mfig = mmax /. figparms;
{{"μ", 0.001, 0.01, 0.1, 0.25}, {"m_max", mfig /. μ → 0.001,
    mfig /. μ → 0.01, mfig /. μ → 0.1, mfig /. μ → 0.25}} // Transpose // MatrixForm
```

Out[268]//MatrixForm=

| $\mu$ | $m_{max}$ |
|---|---|
| 0.001 | 0.0825309 |
| 0.01 | 0.18545 |
| 0.1 | 0.350698 |
| 0.25 | 0.493519 |

And the maximum value attained by E[$\overline{X}$] is

In[269]:=
```
exmax = ex /. m → mmax;
```

It seems that (exmax - p) is positive when $\mu < \mu C$ and negative otherwise, but the formula is too complicated so cannot show this.

In[270]:=
```
LogLinearPlot[(exmax - p)/δ /. {b → 10, c → 1, d → 15, n → 4, p → 0.45, δ → 0.005},
 {μ, 0, 1}, PlotRange → All, ImageSize → Small]
```

Out[270]=



# WF life-cycle, changes with *m*

## Indirect/secondary term

In[235]:=
```
bi = βWFI /. {Qin → QinWF, Qout → QoutWF} /. mychange // FullSimplify
```

Out[235]=
$$\left((-1+\mu)\left((2+d(-2+m))m-2(1+d(-1+m))^2\mu+(1+d(-1+m))^2\mu^2\right)\right)\Big/\left((-1+d)^2(-2+\mu)\mu(-1+(-1+n)(-2+\mu)\mu)-2(-1+d)m(-1+\mu)^2(-1+d(-1+n)(-2+\mu)\mu)+d\,m^2(-1+\mu)^2(-1+d(-1+n)(-2+\mu)\mu)\right)$$

Derivative with respect to *m*

In[236]:=
```
dbi = D[bi, m] // FullSimplify
```

Out[236]=
$$-\left(\left(2(-1+d)^3(1+d(-1+m))n(-2+\mu)^2(-1+\mu)\mu^2\right)\Big/\left((-1+d)^2(-2+\mu)\mu(-1+(-1+n)(-2+\mu)\mu)-2(-1+d)m(-1+\mu)^2(-1+d(-1+n)(-2+\mu)\mu)+d\,m^2(-1+\mu)^2(-1+d(-1+n)(-2+\mu)\mu)\right)^2\right)$$

In[237]:= `Solve[dbi == 0, m] // FullSimplify`

Out[237]= $\left\{\left\{m \to \frac{-1+d}{d}\right\}\right\}$

In[238]:= `dbi /. m → 0 // FullSimplify`

Out[238]= $\frac{2\, n\, (-1+\mu)}{(-1+(-1+n)\, (-2+\mu)\, \mu)^2}$

–> $\beta$WFI decreases until $m = \frac{d-1}{d}$, then increases.

## EX

In[239]:= `ex = EXWF // ApplyParms;`

Derivative with respect to *m*

In[240]:= `dex = D[ex, m] // FullSimplify`

Out[240]= $\left(2\, (-1+d)^3\, (1+d\, (-1+m))\, n\, (-1+p)\, p\, \delta\, (-2+\mu)^2\, (-1+\mu)\, \mu\, (c+b\, (-2+\mu)\, \mu)\right) \Big/ \left((-1+d)^2\, (-2+\mu)\, \mu\, (-1+(-1+n)\, (-2+\mu)\, \mu) - 2\, (-1+d)\, m\, (-1+\mu)^2\, (-1+d\, (-1+n)\, (-2+\mu)\, \mu) + d\, m^2\, (-1+\mu)^2\, (-1+d\, (-1+n)\, (-2+\mu)\, \mu)\right)^2$

In[241]:= `mcc = Solve[dex == 0, m] // FullSimplify`

Out[241]= $\left\{\left\{m \to \frac{-1+d}{d}\right\}\right\}$

Second derivative with respect to *m*, evaluated at the critical *m*

In[242]:= `ddex = D[ex, {m, 2}] /. mcc[[1]];`

When does this second derivative change signs?

In[243]:= `Solve[ddex == 0, μ] // FullSimplify`

Out[243]= $\left\{\{\mu \to 0\}, \{\mu \to 1\}, \{\mu \to 2\}, \{\mu \to 2\}, \left\{\mu \to 1 - \frac{\sqrt{b\, (b-c)}}{b}\right\}, \left\{\mu \to \frac{b+\sqrt{b\, (b-c)}}{b}\right\}\right\}$

It is either a minimum or a maximum depending on $\mu$. To know whether it is a min or a max, let’s look at the initial change of EX with *m*, and see whether it is increasing (-> max) or decreasing (->min).

In[244]:=
```
dex0 = dex /. m → 0 // FullSimplify
Solve[% == 0, μ]
Limit[dex0, μ → 0]
```

Out[244]= $-\frac{2\, n\, (-1+p)\, p\, \delta\, (-1+\mu)\, \left(c+b\, (-2+\mu)\, \mu\right)}{\mu\, \left(-1+(-1+n)\, (-2+\mu)\, \mu\right)^2}$

Out[245]= $\left\{\{\mu \to 1\},\ \left\{\mu \to \frac{b-\sqrt{b^2-b\, c}}{b}\right\},\ \left\{\mu \to \frac{b+\sqrt{b^2-b\, c}}{b}\right\}\right\}$

Out[246]= $n\, (-1+p)\, p\, \delta$ DirectedInfinity[c]

-> – When $\mu < \mu_c^{WF} = 1 - \frac{\sqrt{b\,(b-c)}}{b}$,

$E[\overline{X}]$ reaches a minimum at $m_c^{WF} = \frac{d-1}{d}$, i.e., initially decreases with *m*.;

–Otherwise, $E[\overline{X}]$ reaches a maximum at $m_c^{WF} = \frac{d-1}{d}$, i.e., initially increases with *m*.

With our parameters, the critical value of $\mu$ is $\mu_c^{WF}$=

In[247]:= $1 - \frac{\sqrt{b\,(b-c)}}{b}$ /. figparms // N

Out[247]= 0.0339082